

OXFORD PASCAL

FOR THE COMMODORE 64

LIMBIC SYSTEMS UK LTD

64 PASCAL
USERS MANUAL

By David Goodman
Copyright © 1984 All rights reserved
LIMBIC SYSTEMS UK LTD
FEBRUARY 1984

COPYRIGHT

The materials on this diskette and in the manual are copyrighted by LIMBIC SYSTEMS UK LTD. Copying for resale or exchange is illegal and is strictly forbidden.

WARNING

Although programs are tested by L.S.UK before release, no claim is made concerning the accuracy of this software. L.S.UK and its distributors cannot assume liability or responsibility for any loss or damage arising from the use of these programs, and programs are sold only on the basis of this understanding. Individual applications should be thoroughly tested before implementation, as programs are standardized in order to offer low cost software, and therefore may not necessarily fulfil a particular requirement. You are advised to consult your business software dealer for installation, maintenance, and training costs, should you require these services. Any changes to the software not recommended by L.S.UK may lead to the withdrawal of support services.

DISKETTE CARE

Mini-diskettes appear to be reasonably tough, but are in fact very delicate and require careful handling. They must at all times be either in a disk drive or in their protective envelope. Diskettes left lying about stand a very good chance of never functioning again.

The magnetic surface should never be touched – note that contrary to popular belief, programs and data are recorded on the **UNDERSIDE** of the diskette.

The environment of the diskettes should not have a temperature below 10°C (50°F), or above 52°C (125°F).

The diskette should never be bent or flexed.

Insertion of a diskette into a drive should always be done in a gentle, cautious manner.

Diskettes should always be kept away from all magnetic fields (e.g. Electric motors, high current cables, etc. . .).

CONTENTS

1. INTRODUCTION TO OXFORD PASCAL

2. BEGINNER'S GUIDE TO PASCAL

2.1 Getting Started

- 2.1.1 WRITE statements, strings
- 2.1.2 Integer arithmetic. +, -, *, DIV, MOD
- 2.1.3 Functions: ABS, SQR, ODD
- 2.1.4 Boolean expressions >, <, >=, <=, <>, =, AND, OR, NOT

2.2 Pascal Statements

- 2.2.1 Variables and assignment statements
- 2.2.2 FOR statement
- 2.2.3 IF statement
- 2.2.4 REPEAT and READ statements
- 2.2.5 CASE statement
- 2.2.6 Error messages and error correction
- 2.2.7 WHILE statement

2.3 More Variable Types

- 2.3.1 Real numbers
- 2.3.2 Real arithmetic.
- 2.3.3 SQRT, SIN, ARCTAN, LN, EXP, ROUND, TRUNC.
- 2.3.4 Output formatting and constants
- 2.3.5 Characters
- 2.3.6 64 graphics
- 2.3.7 Arrays
- 2.3.8 Enumerated types and subranges ORD, PRED, SUCC
- 2.3.9 Sets

2.4 Procedures And Functions

- 2.4.1 Procedures
- 2.4.2 VAR parameters
- 2.4.3 Functions
- 2.4.4 Recursion
- 2.4.5 Textfiles
- 2.4.6 Strings

2.5 Advanced Features

- 2.5.1 Records
- 2.5.2 Pointers and lists
- 2.5.3 GOTO statement
- 2.5.4 Extensions

2.6 Disk-based Operation

2.7 Editor Command Summary

2.8 Error Messages

2.9 Sample Programs

3. OXFORD PASCAL REFERENCE MANUAL

3.1 General

- 3.1.1 Pascal Keywords
- 3.1.2 Pascal Identifiers
- 3.1.3 Other Special Symbols
- 3.1.4 Comments
- 3.1.5 Constants
 - integer
 - real
 - character and string
- 3.1.6 Blanks

3.2 Data Types and Operators

- 3.2.1 Integer
- 3.2.2 Real
- 3.2.3 Char
- 3.2.4 User-defined (enumerated) Types
- 3.2.5 Subrange Types
- 3.2.6 Boolean
- 3.2.7 Operator Precedence
- 3.2.8 Summary of Arithmetic and Conversion Functions

3.3 Pascal Declarations and Statements

- 3.3.1 Pascal Programs
 - 3.3.1.1 Constant Declarations
- 3.3.2 Type Declarations
- 3.3.3 Variable Declarations
- 3.3.4 Executable Statements
- 3.3.5 Assignment Statements
- 3.3.6 Compound Statements
- 3.3.7 "If" Statements
- 3.3.8 "Repeat" Statements
- 3.3.9 "While" Statements
- 3.3.10 "For" Statements
- 3.3.11 "Case" Statements
- 3.3.12 "Goto" Statements label declarations

3.4 Input and Output of Text

- 3.4.1 Outputting to Textfiles
- 3.4.2 Inputting from Textfiles
- 3.4.3 Reading Other Data Types from Textfiles
- 3.4.4 Writing Other Data Types to Textfiles
- 3.4.5 Abbreviations
- 3.4.6 Manipulating Files

3.5 Structured Data Types

- 3.5.1 Arrays
- 3.5.2 Sets
- 3.5.3 Records
- 3.5.4 Packed Structures
- 3.5.5 Pack and Unpack

3.6 Functions and Procedures

- 3.6.1 Function and Procedure Definitions
- 3.6.2 Procedure and Function Calls
- 3.6.3 Parameters
- 3.6.4 Local Declarations
- 3.6.5 Recursion and Forward References

3.7 Dynamic Storage and Pointers

- 3.7.1 Pointers
- 3.7.2 "New" and "Dispose"

3.8 Disk Files

- 3.8.1 Declarations
- 3.8.2 Sequential Writing
- 3.8.3 Sequential Reading
- 3.8.4 External Files
- 3.8.5 Reading and Writing from other Devices
 - Disk textfile example
- 3.8.6 CLOSE command

3.9 Extensions to Standard Pascal

- 3.9.1 Hexadecimal Constants
- 3.9.2 Memory VDU and Port Access
- 3.9.3 Added commands for Oxford Pascal V1.0
- 3.9.4 Hexadecimal Input and Output
- 3.9.5 Bit Manipulation
- 3.9.6 Catching I/O Errors
- 3.9.7 Keyboard Interrupts
- 3.9.8 Random Number Generator
- 3.9.9 Underscore
- 3.9.10 The 64 Internal Clock
- 3.9.11 Input of String Variables
- 3.9.12 Program Chaining

3.10 64 Pascal Interface Guide

- 3.10.1 Assembly Language Format
- 3.10.2 Storage Formats

ERRATA

Section	Page
---------	------

1	2	FFF HEX should read FFFF HEX
2	1	OXFORD Pascal CX. should read OXFORD Pascal VX.X
2.2	5	the line :integer; should read x:integer;
2.3	1	writeln(x+y:7:2,x-y:7:2,x*y:7:2 x/y:7:2); should read writeln(x+y:7:2,x-y:7:2,x*y:7:2,x/y:7:2);
2.3	4	read i some numbers should be read in some numbers
2.3	6	The final line in the program Eratosthenes should be end.
2.4	2	The last line of the first example procedure should be end.
2.4	3	, for example input becomes , for example input↑
2.4	4	begin while (input=' ') and not eoln do becomes begin while (input↑=' ') and not eoln do
2.4	4	repeated if eoln then readln; becomes repeat if eoln then readln;
2.4	4	until input=' '; becomes until input↑=' ';
2.4	4	read (x) is equivalent to x:=input;get(input) becomes read(x) is equivalent to x:=input↑;get(input)
2.4	4	write(x) is equivalent to output:=x;put(output) becomes write(x) is equivalent to output↑:=x;put(output)
2.4	4	x:=input becomes x:=input↑
2.5	1	secs=0 then should read if secs=0 then
2.5	1	a line is missing after - for l:=1 to delay do (*nothing*); - which should be secs:=(secs+1)mod60;
2.7	3	the drive defaults to drive 0 becomes the drive defaults to the last drive accessed
2.7	4	The BASIC commands available should include LOAD and PEEK
2.9	1	row:=col+j; should be col:=col+j;
3.2	1	Integer is misspelt as Interger
3.3	3	if y the s1 else s2 becomes if y then s1 else s2
3.4	2	the line integer :reads one character into the variable, as above should be omitted
3.9	1	linefeed = α; becomes linefeed = 'α';
3.9.12	1	onject code in "progl.obj" becomes object code in "progl.obj"

1. INTRODUCTION TO OXFORD PASCAL

Pascal is a powerful high level computer language written by Niklaus Wirth of Zurich, Switzerland.*

It can be efficiently implemented on small computers as well as large mainframes, offering numerous advantages over other popular microcomputer languages such as BASIC. Some of these advantages are:

- ALGOL – like block structure
- Meaningful variable names
- Powerful data structuring techniques
- User-defined data types and constants
- Excellent function and subroutine linkage
- Recursive calls
- Clean, modern flow control
- Runtime error checking
- Dynamic variable allocation
- Greater standardisation
- High speed of execution
- Greater program legibility

OXFORD Pascal is an implementation of standard Pascal designed specially for the CBM 64. It offers all the features of this powerful language together with some useful enhancements for the personal computer user.

OXFORD Pascal has two modes of operation. In the simplest mode the Pascal compiler co-resides in RAM with the user's program. This is ideal for learning the language or writing small programs which do not need the disk. Most Pascal commands are available in this mode except those involving diskette files. For more complex programs the disk-based compiler can be used to give the full power of the language.

* PASCAL USER MANUAL AND REPORT BY JENSEN AND WIRTH
— Springer-Verlag 1975

Hardware requirements

CBM64 computer, plus a model 1541 floppy disk unit or a 4040 floppy disk unit with an INTERPOD from O.C.S.S.

Some implementation information

MAXINT = 32767

type INTEGER = -32768...32767

type CHAR = the ASCII set (extended to include 64 graphics)

set values: must be in 0..127 (therefore set of char must be between chr(0)..chr(127))

real numbers: accuracy: to 9 digits

range: approx 1E-38 to 1E38

default output formats: integer : 7 characters
 real : 12 characters
 boolean : 6 characters
 char : 1 character
 string : size of string

program size and complexity: no restriction, apart from exceeding the total memory capacity of the system (STACK OVERFLOW is printed)

identifiers: first 8 characters must be unique

labels: first 8 digits must be unique

Extensions to standard Pascal

Dynamic specification of filenames
 Input of strings
 Hexadecimal numbers and hex I/O
 Bit manipulation
 Machine language interface
 Memory and VDU screen access
 Run-time I/O error detection
 Random number generator
 Program chaining
 64 clock interface
 Separate compilation (linking)

PASCAL RAM USAGE

This is automatically optimised to use all available memory

FFF HEX	Kernal ROM or graphics bit-map
E000 HEX	I/O drivers or colour graphics memo
D000 HEX	editor
C000 HEX	
Top of RAM	run time stack
	dynamic variables
	executable code (p-code)
800 HEX	edited text

"Stack overflow" is printed when all memory has been exhausted.

2. BEGINNER'S GUIDE TO PASCAL

This section is a straightforward introduction to some of the features of Pascal.

2.1. GETTING STARTED – WRITE STATEMENTS

Turn your computer on. It should be displaying the message.

```
**** COMMODORE 64 BASIC V2 ****
64K RAM SYSTEMS 38911 BASIC BYTES FREE
READY
```

To run Pascal, turn the disk unit on, insert the Pascal diskette into the disk drive (drive 0 for dual disk drives), and then type:

```
LOAD "*",8
```

followed by a carriage return. The computer should reply with:

```
SEARCHING FOR *
LOADING
READY
```

This loads the first program on the diskette into memory, which should be the Pascal system. Now you just type:

```
RUN
```

followed by a carriage return.

You should get a clear screen with the PASCAL SIGN-ON MESSAGE:

```
OXFORD Pascal cX.X
```

```
loading...
ready.
```

Modern computer can be very powerful, but they need to be "told" what to do by means of PROGRAMS. Computers work in a language of numbers called MACHINE LANGUAGE, but machine languages are generally quite difficult for humans to master, and they differ widely depending on the particular computer you are using.

It's much easier to talk to the computer in a HIGH LEVEL LANGUAGE such as COBOL or Pascal. These languages somewhat resemble English, but have stricter rules of grammar to prevent ambiguities.

Pascal was invented by Niklaus Wirth, of Zurich, Switzerland in 1968. (It is named after the 17th century French mathematician Blaise Pascal). Pascal is an ideal language for learning to write computer programs. Your Pascal programs are automatically translated by the computer into a machine language which it can interpret.

Let us start right away with a very simple programming example:

2.1.1 Example 1

First of all you must enter the program into the computer memory, and this is done using the EDITOR.

Type in the first line of the program, shown below. Follow it with the carriage return key (referred to in this manual as <return>):

```
10 begin
20 write ('Hello World!')
30 end.
```

Having typed in the first line, the editor should automatically prompt you with the next line number, which you should not re-type:

20

These line numbers have no significance in Pascal – they are purely for use by the editor, and they will be assumed in all future examples. Remember, if you make a mistake in typing the program you can correct it by using the screen editing commands INST, DEL, cursor up, cursor down, cursor left and cursor right, just as in BASIC. Pressing the DEL key, for example, will erase the last key you typed. See the CBM 64 User's manual for a complete description.

Now enter the remaining two lines of the program, being especially careful about punctuation and spelling, and don't forget that full stop at the end!

```
write ('Hello World')
end.
```

When you have finished, type a blank line (just <return>) to turn off the line numbers.

All you have to do to run your program is to type

r<return>

If all goes well the computer should reply with something like the following:

```
Compiling
Program 0 0909
0 error(s)
Compilation complete.
```

Do not worry about the details, but what is happening is that the computer is scanning your program and converting it into a numeric form which it can efficiently execute. (If you don't get the message: "0 error(s)", then you probably made a mistake in typing. You could try typing "new <return>" and starting again!)

Now the computer should automatically run your program, and print the message:

Hello World

Once your program has been compiled, it can be run as many times as you like by typing:

r<return>

Each time the computer should print:

Hello World

Now let us look at the program in more detail. The main body of a Pascal program is always enclosed between the words BEGIN and END, the final END must be followed by a full stop. Pascal programs consist of a sequence of "statements" which are executed sequentially in the order they are written. Example 1 has one statement, a WRITE statement which tells the computer to write something on the screen, in this case the message "Hello World!". The object enclosed in the single quotes is called a STRING and may contain any sequence of characters except <return>. Also, if a single quote is itself to be included in a string, it should be doubled up, so that the Pascal program

```
begin
write ('O'Brien's string')
end.
```

would cause the message

O'Brien's string

to be printed on the screen.

Example 2

Other things can be printed besides strings. Try the following program. We will use the same steps as example 1 but we must remember to erase example 1 from the computer memory. So type:

new <return>

now type example 2 into the computer:

```
begin
write (3 + 4);
write (6 - 2 - 1)
end.
```

followed by

r<return> (to compile and run the program)

When the program is run the computer should print

7 3

Example 2 contains two statements, which must be separated by a semicolon. It has examples of INTEGER (whole number) arithmetic.

Now try the next example:

2.1.2 Example 3 multiplication and division

```
begin
write (6*7, 18 div 4, 18 mod 4, -(4+2)*3)
end.
```

The computer should print

42 4 2 -18

In Pascal "*" means multiplication, DIV means integer division (ie with rounding towards zero), and "18 MOD 4" gives the remainder when 18 is divided by 4.

Note how brackets have been used to change the order of evaluating $-4 + 6$, or 2. This is because the computer does multiplications and divisions before it does additions and subtractions.

Any number of items can be printed using a single WRITE statement, provided that they are separated by commas.

2.1.3 Example 4 functions

```
begin
write (sqr (4 + 5), abs (- 44), abs (44), odd (3))
end.
```

The computer should print

81 44 44 TRUE

SQR, ABS and ODD are called "functions". There are many different functions in Pascal.

SQR, followed by a number in brackets, gives the square of the number.

ABS gives the absolute value of the number.

ODD (3) is TRUE because 3 is odd.

The last function, ODD, gives a Boolean, or logical result, that is it can either be TRUE or FALSE. Boolean values are used a lot in Pascal so let us look at them more closely.

2.1.4 Example 5 Boolean expressions

```
begin
write (true, false, 3 = 3, 3 = 4);
write (3 <> 4, 5 < 6, 9 >= 10);
end.
```

Should print:

TRUE FALSE TRUE FALSE
TRUE TRUE FALSE

because: 3 is equal to itself
 3 is not equal to 4
 etc.

= means "equal to"
< means "less than"
> means "greater than"
>= means "greater than or equal to"
<= means "less than or equal to"
<> means "not equal to"

WRITELN is like WRITE but also generates a new line after printing all the values in brackets.

Example 6 Boolean expressions

These can get a bit complicated, but the computer evaluates them using the rules of logic.

```
begin
write ((3 = 3) and (3 < 5), (3 = 4) or (3 > 11));
write (not true, not false, not (1 = 2));
end.
```

Gives the result:

TRUE FALSE FALSE TRUE TRUE

because both (3 = 3) and (3 < 5) are true, neither (3 = 4) nor (3 > 11) are true, and (1 = 2) is false so not (1 = 2) is true.

"x and y" is TRUE if both x and y are TRUE

"x or y" is TRUE if either x or y (or both) are TRUE

"not x" is TRUE if x is FALSE, and FALSE if x is TRUE

2.2 PASCAL STATEMENTS

First a word about symbols. These are the building blocks of Pascal programs, and there are three main kinds:

1. Pascal keywords, such as BEGIN and END, which are reserved and can't be altered by the user. A complete list of these is given in the reference manual, section 3.1.1.
2. Special symbols such as `;`, `:=`, `<`, `>` etc.
3. Identifiers, which are names chosen by the user. They can be any sequence of letters or digits, but must start with a letter. For example:

```
i
Henrythe8th
PI
```

Warning

Identifiers are unique only if they differ in the first 8 characters, so that Henrythe7th and Henrythe8th are the same identifier in OXFORD Pascal (and many other implementations).

Upper case letters are equivalent to their lower case counterparts so that PI, pi and Pi are all synonymous.

Some standard identifiers such as WRITE and WRITELN are predeclared in every version of Pascal.

These can be redefined by the user, however (in contrast to Pascal keywords).

Important

Pascal symbols can't contain imbedded blanks. "Henry the 8th" is not the same as "Henrythe8th", and "30 000" is not equivalent to the number 30000. ("30,000" would also be illegal). Note especially that `:` `=` cannot be used instead of `:=`.

This aside, spaces, tabs and new lines may occur anywhere in a Pascal program, and are ignored.

Now we return to some actual examples of Pascal programs. Indentation is used by putting spaces in front of certain lines. This is optional but helps to make the program clearer to humans.

2.2.1 Example 7 Variables and assignment

```
var x,y :integer;
begin
x:=3; y:=27;
writeln(x,y);
x:=4;
y:=x+2;
write(x,y, x+y)
end.
```

Should print:

```
3  27
4  6  10
```

The VAR declaration comes before the BEGIN, and informs the compiler that the identifiers x and y are "variables" which can take integer values. As the name implies, variables can change in value throughout the execution of the program. In line 3, the value of x is set to 3 and the value of y is set to 27. Then later, x is set to 4 and y is set to $x + 2$, or $4 + 2 = 6$. Notice that $y := y + 2$ could also have been written setting y to $27 + 2 = 29$. Variables can also be declared as BOOLEAN and many other types besides INTEGER.

2.2.2 Example 8 Repetition using "FOR" loops

```
var i : integer;
begin
writeln('going up');
for i := 1 to 5 do writeln(i);
writeln('going down');
for i := 5 downto -1 do writeln(i);
end.
```

Should print:

```
going up
1
2
3
4
5
going down
5
4
3
2
1
0
-1
```

The statement following the "FOR . . DO" (in this case a WRITELN statement) is repeated once with each value of the variable i.

2.2.3 Example 9 "if" statements

```
var i : integer;
begin
for i := 1 to 11 do
begin
write(i);
if odd(i) then writeln(' is odd')
else writeln(' is even');
end
end.
```


The result should be:

1 is odd	7 is odd
2 is even	8 is even
3 is odd	9 is odd
4 is even	10 is even
5 is odd	11 is odd
6 is even	

"if" statements give the computer a choice of two statements to do, depending on the value of the Boolean expression. (Remember, Boolean expressions can either be TRUE or FALSE). The "else" part of a conditional statement is optional but — IMPORTANT — "else" is never preceded by a semi-colon.

The WRITE and IF statements in our example are enclosed in BEGIN . . . END to make them act as a single statement to be repeated by the FOR loop.

2.2.4 Example 10 finding the average

This example introduces keyboard input, and a more general sort of loop.

```
var total, count, x : integer;
begin
  total:=0; count:=0;
  write("Type some numbers: ");
  repeat
    read(x);
    total:=total+x;
    if x>0 then count:=count+1;
  until x=0;
  writeln("The average is", total/count);
end.
```

When you run this program, the computer should invite you to type a series of numbers. Try typing:

3 47 5 199 0 <return>

The computer should reply with

The average is 6.35000E+01

The statement read (x) tells the computer to accept an integer from the keyboard and place its value in the variable x. If you type something the computer doesn't recognise as an integer, you might get the message

INTEGER READ ERROR line 60

and the program will terminate.

The "/" operator gives division with a floating point, or REAL result (whereas DIV gives an integer result). More about REAL arithmetic later.

The "real" number was printed in "scientific" notation, which will be familiar to many calculator users. The number following the "E" represents a power of 10, so that 6.35E+01 means "6.35" (times 10 to the power of +1) or 63.5.

The printing format can be changed to make it more legible by specifying the total number of characters you would like printed and the number of digits after the decimal point. (Rounding is done automatically). Thus:

Writeln('The average is', total/count : 10 : 3)

would have printed

The average is 63.500

The number is printed with 4 leading blanks to give the 10-character field you specified.

The "repeat"... "until" loop simply executes the enclosed statements until the condition at the end turns out to be TRUE. In our example the loop is terminated when a zero is read from the keyboard.

2.2.5 Example 11 Case statement

This sample introduces a slightly more elaborate way of choosing one of several statements:

```
var verse, i : integer ;
begin
  for verse:=1 to 4 do
    begin
      writeln;
      for i:= verse downto 0 do
        case i of
          3: writeln('three men');
          2: writeln('two men');
          1: writeln('one man');
          0: writeln('and his dog')
        end
      end
    end
  end
```

This should result in the printout:

one man
and his dog

two men
one man
and his dog

three men
two men
one man
and his dog

case error line 70

The error message was caused because in the last verse i becomes 4 and there is no corresponding label in the CASE statement. Case labels can also be combined, for example

4,5,6: writeln('Many men');

2.2.6 A note on error messages

The "CASE ERROR" message is called a "runtime" error message because it occurs while the program is actually running. There are several such messages which you may encounter (see section 8).

By now you may have started to experiment with your own programs. (This is probably the best way of finding out what is and is not possible in Pascal). If so, you will sooner or later get one of the compiler's error messages. This may also happen if you make a mistake in typing one of the examples. The simple program

```
var x : boolean;  
    : integer;  
begin  
  read (x);  
  write (x)  
end.
```

would cause the following message during compilation:

```
compiling  
---ERROR TYPE 46 LINE 20 NEAR X  
program 0 090d  
1 error(s)  
Compilation complete
```

NOTE---the line number may sometimes be out by 1 line or even more, depending on how long the compiler needed to detect the error.

Typing "L" in response to the prompt:

ready

displays the program on the VDU as well as compiling it. All the line numbers and errors are marked. In our example:

```
10 var x :boolean;  
20  x <- ERROR 46  
20  x : integer;  
30  begin  
40    read (x);  
50    write (x)  
60  end.
```

The full version of line 20 is retyped underneath the error report.

The computer will not let you run a program if there are any compiler errors.

Correcting errors

This can be done using the editor, without having to retype the whole program. To correct the small example above you might type:

list

to list the program on the screen:

```
10 var x : boolean;  
20  x : integer;  
30 begin  
40  read (x);  
50  write (x)  
60 end.
```

To delete the second line, type 20 followed by <return>. Typing "list" again should give:

```
10 var x : boolean;  
30 begin  
40  read (x);  
50  write (x)  
60 end.
```

Line 10 is still wrong. We want to read and write integer, so we type
change /boolean/integer/

which makes the substitution and retypes the line.

"list" now should print the correct version of the program.

```
10 var x : integer;  
30 begin  
40  read (x);  
50  write (x)  
60 end.
```

The program doesn't do much, just reads a number from the keyboard and prints it out again.

For a more complete explanation of how the editor works, see the command summary (section 7). This also explains how to load and save your Pascal program on diskette.

2.2.7 While statement

There is another sort of loop in Pascal, besides REPEAT and FOR loops.

The WHILE statement is like the REPEAT statement except that the test is done at the beginning of the loop (so that the loop need not be executed at all). Also, like the FOR loop only one statement may be repeated (or a sequence of statements enclosed in BEGIN and END).

Example:

```
i:=1;  
while i <= 5 do  
  begin  
    writeln (i);  
    i:=i+1;  
  end;
```

Has the same effect as

```
for i:=1 to 5 do writeln (i);
```


2.3. MORE ABOUT DATA TYPES IN PASCAL

2.3.1 Example 12 Floating point numbers

```
begin
    writeln (3.3, 33.0, 330.0, 0.33);
    writeln (-3.3E3, 3.3E-1, 4.5+2.1)
end.
```

The computer should print

```
3.30000E+00 3.30000E+01 3.30000E+02 3.30000E-01
-3.30000E+03 3.30000E-01 6.60000E+00
```

The presence of either a decimal point or an exponent (the "E" part) in a number tells Pascal to treat it as a floating point or a REAL number.

3.3E3 means 3.3 times (10 to the power of 3)
in other words 3.3 x (10x10x10) or 3300

Floating point numbers in Pascal have an accuracy of 9 digits and may range in size from about 1E-38 to 1E38. In contrast to integer, you should not expect Pascal real arithmetic to be exact. This means, for example, that 4.0 may in fact be printed as 3.999999. Also you can't rely on testing real numbers for equality. 2.0 + 2.0 = 4.0 may not always be true!

2.3.2 Example 13 floating point arithmetic

```
var x, y: real;
begin
    x := 9.1;
    y := 8.7;
    writeln (x+y: 7:2, x-y: 7:2, x*y: 7:2 x/y: 7:2);
    writeln (sqr(x) : 7:2, sqrt(x): 7:2, abs(x): 7:2);
    write (trunc(x), trunc(y), round(x), round(y))
end.
```

Should print:

17.80	0.40	79.17	1.05
82.81	3.02	9.10	
9	8	9	9

We have already met +, -, * and /. They are used to mean, addition, subtraction, multiplication and floating point division (DIV means integer division. DIV and MOD shouldn't be used with reals).

2.3.3 SQR, SIN, ARCTAN, LN, EXP, ROUND, TRUNC

SQR (X) means the square of X
SQRT (X) means the square root of X
ABS (X) gives the absolute value of X
TRUNC (X) gives the integer (whole number) part of X
ROUND (X) rounds X to the nearest integer.

Some other useful mathematical functions are

SIN (X) gives the sine of X (X is in radians)
COS (X) gives the cosine of X (X is in radians)
ARCTAN (X) gives the angle whose tangent is X (in radians)
LN (X) gives the natural logarithm (base e) of X (for X > 0)
EXP (X) gives the number e raised to the xth power.

1 radian = 57.29578 degrees

e = 2.718281

2.3.4 Example 14 Output formatting constants.

```
program waves;
const f1 = 0.5; f2 = 0.05; amplitude = 19;
var x1, x2, y : real;
begin
    x1:=0; x2:=0;
    repeat
        x1:=x1 + f1;
        x2:=x2 + f2;
        y:=sin (x1) * sin (x2) * amplitude;
        writeln ('x' : round (y) + amplitude)
    until false
end.
```

The program should print an amplitude - modulated sine wave.

Because of the REPEAT..UNTIL FALSE loop, example 14 will continue printing almost forever (at least until x1 or x2 becomes too large!) One way of stopping it would be to turn off the power, but if you did that you would lose the program. A better way is simply to press the STOP key.

The computer should print:

BREAK AT LINE xxxxxx

Where xxxxxx is the line it happened to be executing when you pressed STOP. (If this doesn't happen try again)

The "Program" header is optional in OXFORD Pascal and in this case simply serves to give the program a name: WAVES. The name has no significance to the computer; it's merely there as an aid to documentation.

Any text enclosed between the pairs of symbols (* and *) is also ignored by the compiler. This facility can be used to write comments which help human readers to understand the program. Constants, introduced by the keyword CONST, are values which don't change throughout the program. It is an error to use a constant on the left of an assignment statement or as a parameter in the READ statement.

"CONST" declarations are useful for giving names to special values (for example PI = 3.1415926), and they make the program easier to change later. Try using the editor to alter the frequencies f1 and f2 and the amplitude to give different wave patterns in example 14.

2.3.5 Characters

Note how the program uses a field width specification (a colon followed by an integer value) to tell the computer how many characters to allocate to the 'x' when printing. If too many characters are asked for, enough spaces are printed to make up the difference. If not enough are asked for, the string is truncated on the right, for example

```
write ('Hi there' :5)
```

would print:

```
Hi th
```

Numeric values, however, are always printed in full even if too few characters are specified.

2.3.6 Example 15 64 graphics

```
const ncols = 40;
var line, i: integer;
begin
  page;
  for line:=1 to 24 do
    for i:=1 to ncols do
      if odd (line) and odd (i) or
        not odd (line) and not odd (i) then write (chr(177))
    end.
end.
```

This should fill the screen with a pattern. CHARACTERS in Pascal strings of length 1, for example:

```
'x' '?' ""
```

They belong to the data type "Char", which has 256 possible values in OXFORD Pascal, corresponding to the ASCII character set.

The function ord (ch) gives the ASCII integer code (between 0 and 255) for the character ch, while chr (x) gives the character represented by the integer x. So ord ("?")=63 and correspondingly chr(63)="?".

Note – On the 64, the OXFORD Pascal data type "char" has been extended to the range 0..255 to allow the 64 graphics font to be used. Two of these characters were used in the program above. Try writing programs to give different patterns, using the available characters. A simple change would be replacing 177 and 178 by 173 and 174. The statement "page" simply clears the vdu screen.

2.3.7. Example 16 Arrays

Suppose you wanted to read i some numbers and print them out in reverse order. You would have to store the numbers somewhere because you can't start printing until the last number has been read. If you knew that there were always going to be three values, you could write:

```
var x1, x2, x3 : integer;
begin
  write ('Type 3 number : ');
  read (x1, x2, x3);
  writeln (x3);
  writeln (x2);
  writeln (x1)
end.
```

But for 50 values this would get a bit tedious!

The answer is to use an array variable:

```
const n=3;
var x : array [1..n] of integer;
    i : integer;
begin
  write ('Type ', n:1, ' numbers: ');
  for i:=1 to n do read (x[i]);
  for i:=n downto 1 do writeln (x[i])
end.
```

Running the program and typing the data:

```
463    79    980
```

Should give the result:

```
980
 79
463
```

The declaration of x really declares n variables which can be referred to by giving an index in square brackets. The elements of the array x are thus x[1], x[2], ..., x[n].

The constant n was used so that the number of values read in by the program can easily be changed by altering just one line.

Array elements can be any valid Pascal data type, including including another array. This allows two dimensional (or indeed any dimensional) arrays, and a chess-board for example may be represented as:

```
var chessboard: array [1..8] of array [1..8] of chesspiece;
```

Where chesspiece is some suitable data type, probably a user defined type (more about this later). The 5th square of the 3rd row of the chess-board could then be referred to as:

```
Chessboard [3] [5]
```


Because arrays of arrays are used often in Pascal programs, the abbreviation "chessboard [3,5]" is allowed, and similarly in the declaration:

```
var chessboard : array [1..8, 1..8] of chesspiece;
```

This can be extended to arrays of any dimension.

2.3.8. Defining your own data type

None of the data types so far mentioned (integer, real, boolean, or even char) would be really suitable for describing a piece on a chess-board, so Pascal lets you define your own. This may be done in TYPE DECLARATION, for example:

```
type chesspiece = (pawn, knight, bishop, rook, queen, king);
```

Then a variable of type CHESSPIECE could take any of these six values for example:

```
var mypiece, yourpiece:chesspiece
begin
  :
  :
  mypiece:=rook;
  yourpiece:=queen;
  :
```

Type declarations come after constant declarations and before variable declarations. The identifiers used in an 'enumerated' data type like CHESSPIECE must be unique, they can't appear in other enumerated types or be declared as constants or variables. Enumerated types are ordered so that our chess pieces can be compared using =, > etc:

```
king>queen
queen>rook
```

and so on:

Three functions are also defined: PRED, SUCC and ORD

pred (x) gives the value preceding x.

succ (x) gives the value succeeding x.

ord (x) gives the position of x within the data type.
(starting with pawn = 0)

so pred (bishop) = knight

succ (rook) = queen

but pred (pawn) and succ (king) are both meaningless

ord (knight) = 1

ord (rook) = 3, and so on.

2.3.9 Example 17 The sieve of Eratosthenes

This program finds and prints all the prime numbers between 2 and 127.

```
program Eratosthenes;
const n=127;
var sieve : set of 2..n;
    number, i : integer;
begin
  sieve := [2..n];
  for number := 2 to n do if number in sieve then
    begin
      writeln (number);
      for i := 2 to n div number do
        sieve := sieve - [i*number]
      end
    end
  end
```

A prime number is divided only by itself and 1. Our "sieve" used for finding the prime numbers, is a new type of variable called a SET variable.

Sets in Pascal are collections of objects enclosed in square brackets. Either an object is in a set or it is not, so:

```
[1,2,3]
[2,3,1]
and [1,1,3,3,2] are all equivalent.
```

The abbreviation x..y in a set means all the items between x and y inclusive, so:

```
[1..4, 10] = [1,2,3,4, 10]
```

We can test whether an item is in a set by using the operator IN. Thus "4 in [1..5]" will give the Boolean result: TRUE.

The type of a set can be any scalar type (ie no an array or a set) except REAL. Values are restricted to the range 0..127 (so "set of char" from chr(0) to chr(127)) is acceptable).

Now back to our sieve program. Starting with the number 2 and working upwards, if a number is still in the sieve then it's a prime. We simply eliminate all multiples of that number from the sieve because they are not prime. Operations allowed on two sets x and y are:

x + y	which gives the set of all items present in either x or y or both.
x - y	which gives all items present in x which are not also in y.
x * y	gives all the items present in x and also present in y.
x = y	tests if two sets are equal.
x <> y	tests if two sets are not equal.
x <= y	tests if all items in x are also in y.
x >= y	tests if all items in y are also in x.

2.4 PROCEDURES AND FUNCTIONS

2.4.1 Example 18 procedures

```
var ch: char;
procedure lineof (wotsit : char);
  var i: integer;
  begin
    for i:=1 to 30 do write (wotsit);
    writeln
  end; (* of procedure "lineof" *)
begin (* of main program *)
  lineof('?');
  writeln;
  for ch:= 'a' to 'f' do lineof(ch)
end.
```

you don't need to type the comments (* ...*) if you don't want to. These are there to help explain the program.

The computer should print:

```
????? .....
aaaaaa .....
bbbbbb .....
cccccc .....
dddddd .....
eeeeee .....
fffff .....
```

Procedures are used to separate sections of code from the main program, either to make what the program does clearer by dividing it up functionally, or to allow the same code to be "called" from various parts of the program. The procedure "lineof" has a PARAMETER "wotsit" (which takes the data type CHAR). When lineof is called it must be followed by a corresponding actual parameter in brackets. Then "lineof" simply writes a line of wotsit's on the screen.

If a procedure has no parameter then the brackets are omitted. The variable I is "local" to the procedure lineof, the main program doesn't know about it. However lineof could if necessary access the 'global' variable CH. Using local variables helps conserve storage, since they are destroyed when the procedure finishes. Procedures are really mini-programs in their own right. They can have their own constant and data type declarations and even their own procedures.

"WOTSIT" is called a VALUE parameter because a value is substituted for it when the procedure is called. Lineof could change the value of wotsit without affecting the main without affecting the main program. VARIABLE parameters on the other hand are substituted with variables when the procedure is called.

2.4.2 Example 19 Variable parameters

```
var x,y : integer;
procedure swap (var a,b : integer);
  var temp: integer;
  begin
    temp:=a;
    a:=b;
    b:=temp
  end;
begin
  x:=4; y:=77;
  writeln (x,y);
  swap (x,y);
  writeln (x,y)
end
```

This should give the result

```
4    77
77   4
```

Note that it is alright to have local variables, constants and parameters with the same names used in the main program. For example:

```
procedure swap (var x,y : integer);
```

The computer won't get confused (but you might!). The variable parameters a and b are used by SWAP as a means of returning a result to the main program. Another way of returning a value is to define a function.

2.4.3 Example 20 Defining a function

```
var i : integer;
function cube (x : integer) : integer;
  begin
    cube := x*x*x;
  end;
begin
  for i:= 1 to 20 do writeln ("The cube of,
                             i : 2, 'is', cube (i))
end.
```

The program should print some numbers and their cubes. Apart from having to specify a return value, functions are just like procedures.

2.4.4 Example 21 Recursion

A recursive function or procedure is one that calls itself. Using recursion can give neat solutions to mind-bending problems like the "Towers of Hanoi". In this well known puzzle, there are three piles of discs. To start with piles 2 and 3 are empty, and the first pile has a number of discs stacked in order of size, smallest at the top. The game is to get all the discs in the same order (smallest on top) over to the 3rd pile, moving only one at a time, with no disc ever resting on a smaller disc.

```
program Hanoi;
var ndiscs: integer;
procedure move (source, destn, spare: 1..3; n: integer);
begin
    if n > 1 then move (source, spare, destn, n-1);
    writeln('Moving from', source : 2, ' to ', destn : 2);
    if n > 1 then move (spare, destn, source, n-1)
end;
begin
    write ('How many discs ? ');
    read (ndiscs);
    writeln;
    move (1,3,2,ndiscs)
end.
```

Moving one disc is trivial. To move n discs we first move the top ($n-1$) to the spare pile and then move the bottom one. Then the top ($n-1$) are moved using the same technique.

Recursive programs are not always the most efficient, though. They tend to gobble up memory because the computer has to save the variables for each call on the stack. If you make $ndiscs$ too large the computer will run out of memory and print STACK OVRFLOW - line xxxx. The same will happen if you declare more variables in a program than you have memory available, or if you try to compile too large a program.

2.4.5 Text Files

Text files are special Pascal variables having the data type TEXT which are essentially streams of characters with no fixed size. Two are preassigned in OXFORD Pascal. "Input" and "output" are associated normally with the keyboard and the VDU display respectively.

By default, "Input" is implied in READ, READLN, EOLN and EOF and "Output" is implied in WRITE, WRITELN and PAGE. So for example,

EOF is really short for EOF (INPUT)
WRITELN ('Hi!') is really short for WRITELN (OUTPUT, 'Hi')

Each textfile has an associated buffer variable of type CHAR (the file name followed by an upward arrow), for example: input↑

The procedure call:

get (input) reads the next character from the keyboard and puts it in the variable input.

put (output) writes the contents of output to the VDU. So if X is a character variable:

read (x) is equivalent to $x := \text{input}$; get (input)
write (x) is equivalent to $\text{output} := x$; put (output)

Newlines are special character in textfiles. When a file buffer contains one, assignments like

$x := \text{input}$

will set x to a space. Also, the end of line function EOLN will return TRUE.

READLN is like READ but afterwards skips to the beginning of the next line by doing:

```
while not eoln do get (input);
get (input);
```

2.4.6. Example 22 strings

```
program revwords;
const linesize = 64;
type string = packed array [1..LINESIZE] of char;
var word: string;
    nchars, i: integer;
```

```
procedure skipblanks;
begin while (input = ' ') and not eoln do
    get (input) end;
```

```
procedure swap (var s: string; i, j: integer);
var temp: char;
begin
    temp := S[i];
    S[i] := S[j];
    S[j] := temp
end;
```

```
begin
    repeated if eoln then readln;
    skipblanks;
    while not eoln do
        begin
            nchars := 0;
            repeat
                nchars := nchars + 1;
                read (word [nchars]);
            until input = ' ';
            for i := 1 to nchars div 2 do
                swap (word, i, nchars - i + 1);
            write (word: nchars, ' ');
            skipblanks
        end;
        writeln;
    until false
end.
```

The program reads words and writes them out with the letters reversed. For example:

Mary had a little lamb <return>

would print,

ylraM dah a eltil lbma

To stop the program, hit <stop>.

Strings of size n in Pascal are treated as "packed array [1.. n] of char". Packed arrays are like ordinary arrays but are compressed to optimize storage. Packed array ELEMENTS can't be used directly as VAR - parameters (but whole arrays can, as in the example).

Examples of string operations in Pascal:

```
var str: packed array [1..4] of char;
begin
    str := 'when';
    writeln (str > 'what')
end.
```

Would print TRUE because "when" is greater than "what" (Lexicographically, i.e. dictionary order).

2.5 ADVANCED FEATURES

2.5.1 Example 23 Records

```
Program clock;
const delay = 1600; (*approx, for 64*)
var i: integer;
    clock: record
        hrs: 0..23;
        mins, secs: 0..59;
    end;
begin
    write ('Enter the time, in hours minutes seconds : ');
    read (clock.hrs, clock.mins, clock.secs);
    with clock do repeat
        for i:= 1 to delay do (*nothing*);
        secs:=0 then
            begin
                mins:=(mins+1)mod 60;
                if mins=0 then
                    hrs:=(hrs+1)mod 24;
                end;
                writeln (hrs:1,':',mins:1,':',secs:1)
            until false
    end.
```

The program should print out the time roughly every second, for example:

Enter the time, in hours minutes seconds : 1 39 56

Should print:

```
1 : 39 : 57
1 : 39 : 58
1 : 39 : 59
1 : 40 : 0
etc
```

This example is intended to be demonstration of the use of records in Pascal, not a replacement for the 64 built-in clock! See the Reference Manual (3.9.10) for how to access the clock.

Records are a way of combining several conceptually related variables into one structure. The record can then be treated as a whole or the parts can be accessed individually using the *dot* notion.

The WITH..DO statement tells the computer to treat the elements of "clock" as though they were locally defined individual variables for that statement, removing the need for the "clock." prefix.

Record elements can be any type (for example other records or arrays), and in addition an optional "variant" part is allowed (see reference manual).

2.5.2 Example 24 Pointers

```
var p,q: integer;
begin
  new (p); new (q);
  p :=3;
  q :=4;
  write (p ,q )
end.
```

Should print

3 4

The variables p and q are not integers but 'pointers' to integer variables. The actual space for the variables to be stored at is created "dynamically" (in other words while the program is running) by the procedure NEW. This allows programs to create variables as required. A major use of pointers is in processing linked lists:

Example 25 Reversing a line of characters using a list.

```
program revchars;
type itempointer = item;
  item = record
    value:char;
    next: itempointer
  end;
var list,p: itempointer;
begin
  list:=nil;
  repeat
    new (p);
    read (p .value);
    p .next:=list;
    list:=p
  until eoln;
  repeat
    write (p .value);
    p:=p .next
  until p=nil
end.
```

The input: Mary had a little lamb.

Should give the result: bmal eltil a dah yraM

This program defines a record containing a pointer to itself. (A recursive definition). Linked lists give very flexible storage but you have to keep careful track of what points to what.

In standard Pascal the procedure DISPOSE (P) releases the storage assigned to the pointer p and can be used when p is no longer needed.

In Resident mode OXFORD Pascal, dispose has no effect. (However, it is usually possible for programs themselves to implement some sort of "free list" of unwanted items). The Pascal keyword "nil" is a pointer value which points to no variable.

2.5.3 Example 26 "goto" statements

```
label 294, 33;
begin
  33: writeln ("This should be printed");
      goto 294;
      writeln ("This shouldn't");
  294: writeln ("Stuck in a loop");
      goto 33
end.
```

This should print:

This should be printed
Stuck in a loop
This should be printed
Stuck in a loop
etc.

"labels" used with goto statements must be integers and should be declared before constants, data types and variables. GOTO'S should be avoided where possible because they destroy the structure of the program. A common use, however, is for "disaster" exits from nested procedures or statements. Jumping INTO a loop or a procedure will cause unpredictable results.

2.5.4 Extensions to standard Pascal

These are described in the Reference Manual (3.10). One useful procedure is VDU:

Example 27 "poking the vdu screen

```
var i: integer;
begin page;
  for i:=1 to 40 do vdu (i mod 4,i,'x')
end.
```

This should produce a pattern of x's on the screen. Vdu (i,j,ch) stores the character ch at row i, column j.

Remember, PAGE clears the VDU screen.

2.6 DISK BASED OPERATION

So far this manual has been concerned only with using the resident compiler, which is always in RAM. While this may provide an ideal environment for learning Pascal, it necessarily restricts the number of commands available, and the space remaining for user programs.

As you become familiar with Pascal, you will probably want to write larger programs. Using the disk-based compiler and linker, Pascal programs of 6000 lines or more may be run, and this may be extended even further by using program chaining.

You may also want to write programs which access diskette files. While this is possible in resident mode by opening channels to the disk unit, in disk mode the full Pascal file syntax is available, permitting files of any data type.

Disk mode is entered automatically by typing:

DISK

This removes the resident compiler from memory, making the space available for editing.

Once your program is edited, it MUST be saved on disk, for example:

put prog or put 0:prog

saves it in a file called "prog" on drive 0. To compile the program type:

comp prog

The compiler output should be something like:

Insert your system disk followed by a <return>

loading compiler...

Insert your data disk followed by a <return>

Pascal compiler vx.x

program 0 0009

0 error(s)

compilation complete.

There are various compiler options for generating listings etc. If all goes well the compiler will put the object code in a file "prog.obj" which can then be executed. If there are any errors during compilation then the Pascal source must be corrected using the editor and re-compiled. (Use the command "get prog" to read your Pascal text back into memory).

Finally, type:

ex prog

to execute the object file (prog.obj).

The name of each procedure or function is printed out as it is compiled, together with its static nesting level (0 for the main program, 1 for outer level functions and procedures, and so on). A hexadecimal address is also printed, giving a rough idea of its relative position in memory.

The following table summarises the differences between resident and disk mode:

Resident Mode

Compiler always in RAM

Pascal source and object code in RAM

Language differences (see Reference Manual for details):

Textfiles only

DISPOSE is a no-op

Disk Mode

Compiler only in RAM during
a compilation

Source and object code
held in disk files

All file types supported

Disk files fully supported

PACK, UNPACK implemented

DISPOSE fully implemented

Program chaining allowed

2.7 EDITOR COMMAND SUMMARY

Line numbers

A command beginning with a number is recognised by the editor as a new program line, and is inserted in the program text in the position corresponding to that number.

For example:

```
10 end.  
5 begin (* this command comes first *)
```

A line containing just a number has the effect of deleting that line from the program.

Auto

Enables or disables automatic generation of line numbers.

Examples:

```
auto 20 <return> - enables auto numbering with an increment of 20.  
auto <return> - disables auto line numbering.  
default - auto 10.
```

List

Lists the program currently in memory.

Examples:

```
list 100-200 - lists entire program  
list 330      - lists line 330 only  
list 100-     - lists lines 100 onwards  
list 100-200 - lists lines 100 through 200  
list -200     - lists up to and including line 200
```

Note: The STOP key pressed during a listing will halt it completely, while pressing any other key freezes the listing until a second key is pressed.

Upper, Lower

Set upper or lower case display mode (The contents of the program memory are unaffected). Default: lower.

Switching between upper and lower case can also be performed alternately by holding down the shift key whilst depressing the CBM key.

Basic

Return to BASIC, reverting to upper case display.

New

Erase the program memory.

Disk

Enter disk compiler mode.

Resident

Re-load the resident compiler.

Number

Renummer the program lines in memory.
for example:

```
number 1000,2000,30
```

renumbers lines 1000 onwards, starting the new numbers at 2000 and with an increment of 30.

Note that the new starting number must be greater than or equal to the old starting number.

Find

Find and print occurrences of a string in the program.

```
find /function/ - finds all occurrences of "function"
```

```
find /function/,100-250 - finds all occurrences in lines 100 through 250.
```

The "/" can in fact be any delimiting character not contained in the search string.

See note under "list" command to halt execution.

Change

As find, but substitutes a second string for all occurrences found of the first. For example:

```
change/function/procedure/,150
```

Changes all occurrences of "function" in line 150 to "procedure".

See note under "list" to halt command execution.

Delete

Deletes program lines (parameters as with LIST). "Delete" with no parameter is equivalent to NEW.

Put

Saves the program on diskette. For example:

put 0:sara—saves in a file called "SARA" on drive 0. (drive must be initialised).

put @1:jim—saves in an existing file called "JIM" on disk 1.

(drive must be initialised).

Note: the drive defaults to drive 0.

Get

Reads a program from diskette. For example

get sara (searches both disks if required).

R (or Run) Resident mode only).

Run the program in memory (compiling first if necessary).

L (Resident mode) – compile, and display the program on the VDU.

P (Resident mode) – compile and list on the printer.

COMP (Disk mode) – compile a program

comp sara – compiles file "sara" giving a relocatable object file "0:SARA.OBJ"

comp sara,l – "l" option gives a listing on the VDU.

Other options: P – list on printer

N – no object code

C – no range checking or line numbers in the object file
(giving slightly faster and more compact code)

l – object file on drive 1

Ex (Disk mode) – execute an object file

ex sara – executes SARA.OBJ

Note – COMP and EX both clear the text buffer.

Set sets the printer device number, printer type (Ascii / 64), auto line feed flag and is menu driven.

Hex

converts from decimal to hexadecimal.

hex 32 <return> gives the result 0020

Decimal

convert from hexadecimal to decimal.

decimal 7f <return> gives the result 127

Dump

list a program on the printer

The DUMP command has the same syntax as LIST.

Cold

cold-start 64 BASIC.

BASIC commands

Any of the BASIC direct-mode commands below may be used in the editor:

PRINT (or ?)

PRINT #

OPEN

CLOSE

CMD

POKE

SYS

FOR

LET

examples:

let ti\$ = "120000"

?ti\$

?fre (0)

for i=1 to 20: ?i, i*i: next i

– set clock to mid-day

– print the time

– print the number of bytes free

Link (disk mode)

For large programs it is desirable, (and often physically necessary) to have some form of modularization. Several Pascal source files with inter dependent functions and procedures may be compiled separately and their object files later "linked" into one file. The linker may also be used to produce directly executable 64 files.

Examples:

link 0:prog=myprog,yourprog,anyprog

links the files MYPROG.OBJ, YOURPROG.OBJ and ANYPROG.OBJ

into one object file PROG.OBJ on disk drive 0.

Note – "link" clears the text buffer.

Restrictions

(a) The program being linked must have indential variable declarations at the outer program level.

(b) Each **outer-level** function or procedure may only be defined in one file.

(c) If the other files need to refer to this function or procedure, a duplicate header should be included, with the body replaced by the keyword "extern".

(d) The first file in the list is assumed to contain the main program. (The other files would normally just contain a dummy main program:

begin

end.)

Linker example:

file f1:

```
program test (input, output);
var i: integer;
procedure x; extern; (* x is defined in the other file *)
procedure y;
begin
  write (i)
end;
begin (* main program *)
  x
end.
```

file f2:

```
program testpart2(input, output);
var i: integer; (* var's must be identical to f1 *)
procedure y; extern; (* y is defined in f1 *)
procedure x;
begin
  i:=3;
  write ('three =');y;
end;
begin
end.
```

The command sequence might be

```
comp f1
comp f2
link 0:test=f1,f2
ex test
```

The program should print "three = 3"

Including other files in a compilation (disk mode only)

If the character "#", followed immediately by a diskette file name, is placed at the **beginning** of a Pascal source line, then this indicates to the compiler that the contents of the specified file are to be included at that point in the program.

This is extremely useful when program segments are to be linked, as global declarations (which need to be the same in each segment) can be kept in a separate file thus simplifying any alterations.

The facility cannot be nested (the included file must itself contain no #filenames).

Locate (disk mode) – makes an acceptable file which can then be loaded under BASIC and executed by just typing "RUN".

example: locate 0:xyz=jane

Creates an executable file xyz on drive 0 from JANE.OBJ

NOTE – "locate" clears the text buffer.

2.8 ERROR MESSAGES

? Syntax error – editor command is mis-spelled or has invalid parameters.

? Out of memory error – there is insufficient memory left to do the command you specified, for example inserting a new program line or "GETTING" a file.

? Illegal quantity error – bad numeric input to an editor command, for example "NUMBER".

? File data error – one of the Pascal library files is not present on the disk or else has been corrupted.

Compiler not resident – The L,P and R commands may only be used in resident mode.

No source program – You typed L or R with no program text present in the computer's memory.

RUNTIME ERRORS

1. STACK OVERFLOW
 - (during compilation) program is too big
 - (during execution) program needs too much variable space or uses too many levels of recursion.
2. INTEGER READ ERROR
 - an integer was expected from the keyboard.
3. INTEGER OVERFLOW
 - overflow when multiplying two integers, or DIViding or MODing by zero, or TRUNCating or ROUNDing too large a number.
4. ARRAY INDEX ERROR
 - an expression used to index an array is outside the declared range.
5. VARIABLE OUT OF RANGE
 - a variable, or a procedure or function parameter has been given a value outside the allowed range for that data type.
6. CASE ERROR
 - there is no label in a case statement corresponding to the value of the selection expression.
7. BAD PCODE
 - your program has been corrupted, or (hopefully not) a system bug. Occurring at random, this may indicate a memory fault.
8. SET VALUE ERROR
 - a set element has gone outside the range 0..127
9. FLOATING POINT OVERFLOW
 - may occur if the result of $+ - x / \text{SQR}$ or EXP is too large.
10. FLOATING POINT READ ERROR
 - a floating point constant was expected from the keyboard.
11. UNDEFINED GOTO
 - a GOTO statement referenced a non-existent label.
12. COMPLEX LOG OR SQUARE ROOT
 - attempt to take the log or square root of a negative number, or the log of zero.

- 13. FILE NOT OPEN FOR READING – READ or GET without a reset first.
- 14. FILE NOT OPEN FOR WRITING – WRITE or PUT without a rewrite first.
- 15. END OF FILE –attempt to read a file with EOF true.
- 16. NO FREE I/O CHANNELS – 64 operating system only allows ten files to be open at one time.
- 17. DEVICE READ ERROR – Bad status byte encountered while reading data from the IEEE bus.
- 20 to 72. DISK ERROR – An error status has been detected on the floppy disk unit. Returns the error type and if possible the offending filename.

2.9 SAMPLE PROGRAMS

Example 1

The character 'o' should appear to "bounce" around the VDU screen. As a variation, try deleting line 13 to produce a pattern on the screen.

```

program bounce (input,output);
const thecowscomehome = false;
      DELAY = 30;
var row, col, i, j, d : integer;
begin
  row := 0;
  col := 0;
  i := 1; j := 1;
page;
repeat
  for d := 1 to DELAY do;
    vdu (row, col, ' ');
    row := col+j;
    row := row+i;
    if (row > 23) or (row < 0) then begin
      begin
        i := -i;
        row := row+i+i;
      end;
    if (col > 39) or (col < 0) then
      begin
        j := -j;
        col := col+j+j;
      end;
    vdu (row, col, 'o');
    until thecowscomehome
  end.

```


Example 2 The game of Nim

```
program nim;
const NROWS = 24;
      delay = 1000;
      coin = 168;
var pile : array [1..3] of 0..NROWS;
      move : record
          ntaken, pileno : integer
      end;
      i : integer;
      key : char;
function gameover : boolean;
begin gameover := (pile[1] + pile[2] + pile[3] = 0) end;

function asc (n : integer) : char;
begin asc := chr(n + ord('0')) end;
procedure display;
      var p, row, col, firstcol : integer;
begin
      page;
      for p := 1 to 3 do
          begin
              firstcol := p*10;
              for row := 0 to NROWS-1 do
                  if pile[p] >= NROWS-row then
                      for col := firstcol + 3 to
                          firstcol+5 do
                          vdu (row, col, chr(COIN));
                  if pile[p] >= then
                      vdu (NROWS-1, firstcol, asc (pile[p] div 10));
                      vdu (NROWS-1, firstcol+1, asc (pile[p] mod 10));
                  end
              end
          end
      end;
```

```
procedure signon;
begin
      page;
      writeln (' *** NIM ***');
      writeln;
      writeln;
      writeln ('I will set up three piles of coins ');
      writeln ('To move, take any number of coins away');
      writeln ('from any pile. The player who clears');
      writeln ('the screen wins. ');
      writeln;
      write (' Now hit any key to start : ');
      while getkey = chr(0) do;
      end;
```

```
Procedure hismove;
      var ok : boolean;
      begin
          writeln ('Now enter your move :');
          with move do repeat
              writeln;
              write ('Pile (1,2 or 3)? ');
              read (pileno);
              ok := pileno in [1..3];
              if ok then
                  begin
                      write ('Number to take away? ');
                      read (ntaken);
                      ok := ntaken in [1..pile [pileno]];
                      end;
                  if not ok then writeln ('What ??');
              until ok;
              with move do pile [pileno] := pile [pileno]
                  - ntaken;
          end; (* of hismove *)
```

```

Procedure mymove;
var bit : array [1..3, 1..4] of boolean;
    parity : array [1..4] of boolean;
    firstbit, x, i, j : integer;
begin
    for i := 1 to 3 do
        begin
            x := pile          for i := pile [i];
            for j := 4 downto 1 do
                begin
                    bit [i, j] := odd (x);
                    x := x div 2;
                end;
            end;
        end;
    for i := 1 to 4 do parity [i] :=
        bit [1, i] <> (bit [2, i] <> [3, i]);
    move.pileno := 1;
    move.ntaken := 0;
    with move do
        if not parity [1] or parity [2] or parity
            [3] or parity [4]) then
            begin
                while pile [pileno] = 0 do pileno
                    := pileno + 1;
                if pile [pileno] = 1 then ntaken := 1
                else
                    ntaken := random mod (pile [pileno]-1)+1
                end
            end
        else begin
            firstbit := 1;
            while not parity [firstbit] do
                firstbit := firstbit + 1;
            while not bit [pileno, firstbit] do
                pileno := pileno + 1;
            for i := firstbit to 4 do
                begin
                    x := 1;
                    for j := 3 downto i do x := x*2;

                    if parity [i] then
                        if bit [pileno, i] then ntaken

                            := ntaken + x
                        else ntaken := ntaken - x;
                    end
                end;
            end;
            with move do pile [pileno] := pile [pileno]
                - ntaken;
        end; (* of mymove *)
end;

```

```

begin
signon;
repeat
    for i:= 1 to 3 do pile [i] := random mod 10 + 6;
    display;
    repeat
        hismove;
        if gameover then writeln ('Congratulations ... You win!')
        else begin
            display;
            mymove;
            for i := 1 to delay do;
            display;
            writeln ('My move was ', move.ntaken
                :3, ' from pile', move.pileno :2);
            if gameover then writeln ('*** I win. ');
            writeln;
        end;
        until gameover;
        write ('Another game ? ');
        while input = '' do get (input);
        read (key);
        while not eoln do get (input);
    until key = 'n';
    page;
end.

```


3. OXFORD PASCAL REFERENCE MANUAL

This manual is intended to be used for quick reference by those familiar with Pascal or a similar programming language.

3.1 GENERAL

3.1.1 Pascal keywords

These are reserved words in Pascal and cannot be redefined. They must be written without embedded spaces or newlines. A complete list is:

and	do	function	nil	program	type
array	downto	goto	not	record	until
begin	else	if	of	repeat	var
case	end	in	or	set	while
const	file	label	packed	then	with
div	for	mod	procedure	to	

3.1.2 Pascal identifiers

These are names chosen by the programmer for variables, constants etc., and should consist of at least one letter, followed by zero or more letters or digits. Upper and lower case letters are equivalent. Identifiers should be unique in the first 8 characters, and must not contain embedded blanks.

The following identifiers are standard (but may be redefined):

abs	eoln	new	read	sqrt
arctan	exp	odd	readln	succ
boolean	false	ord	real	text
char	get	output	reset	true
chr	integer	pack	rewrite	trunc
cos	input	page	round	unpack
dispose	in	pred	sin	write
eof	maxint	put	sqr	writeln

(see also section 9 – extensions).

3.1.3 Other Special symbols

+	<	'(apostrophe) [:=
-	<=	.	;
*	>=	..	(
/	>	(*	,
=	<>	*)	:

These symbols should not contain embedded blanks.

3.1.4 Comments

Pascal comments are enclosed by the symbols (* and *).

Comments are totally ignored by the compiler. They can contain any characters except the closing delimiter "*)".

3.1.5 Constants

Integer Constants

These consist of a sequence of digits, for example:

33 0001 0

No check is made to ensure that the value is less than $2^{**}15$. Integer constants must not contain embedded blanks or commas (see also section 9.1 on hex constants).

Real constants

These are of the form:

<integer part> . <fractional part>
or <integer part> E <exponent>
or <integer part> . <fractional part> E <exponent>

The integer and fractional parts are non-null strings of digits. The "E" may be in upper or lower case in OXFORD Pascal. The exponent is a digit string which may be preceded by a sign [+ or -].

Real constants must not contain ANY embedded blanks.

Examples:

3,14159 4E-9 -387.4E11
1E+30

A real constant which is out of range (greater than about 1E38) will cause an error.

Character and string constants

These are enclosed in single quotes, and may contain any character except a newline. Single quotes are included in a string by writing them twice.

Examples:

'c', '\$', ''' (character constants)
'Hi there!', 'Fred's string' (string constants)
(see also section 9.1 on hex constants).

3.1.6 Blanks

Any number of spaces or newlines may separate two keywords, identifiers, constants or other symbols, but at least one blank is required between adjacent keywords, identifiers and numbers.

3.2 DATA TYPES AND OPERATORS

3.2.1 Integer

Pascal integers are whole numbers in the range $- \text{MAXINT}$ to $+ \text{MAXINT}$ where MAXINT is an implementation defined constant (32767 in OXFORD Pascal).

OXFORD Pascal stores integers in 16-bit 2's complement form, so integers may range from -32768 to $+32767$.

Integer operators are

- $+$ addition
- $-$ subtraction
- $*$ multiplication
- div integer division (result is rounded towards zero)
- mod remainder operator
- $-$ (unary operator) negation

$+$ and $-$ complement results mod $2^{**}16$.

$*$, div and mod are defined only on values in the range $-\text{MAXINT}.. \text{MAXINT}$, and the result must be in this range (otherwise an error occurs).

Division by zero causes an error.

$x \text{ mod } y = x - ((x \text{ div } y) * y)$

3.2.2 Real

Real numbers in OXFORD Pascal are held in floating point binary form with a 32-bit mantissa (9 digits). The exponent can range from -38 to $+38$.

The operators $+$, $-$, $*$ behave as for integers, but produce a REAL result. (Which will cause an error if it is out of range).

The operator $/$ denotes floating point division. Division by zero will cause an error.

Integer expressions and constants can be used wherever a real expression is acceptable, but real values can't be used with DIV or MOD .

Conversion from real to integer is done by the functions TRUNC and ROUND (sections 2.8).

3.2.3 Char

The Pascal data type "char" operates on an ordered set of characters. In OXFORD Pascal the 128 character ASCII set is used. (Extended to 256 characters to include 64 graphics).

In all implementations of Pascal the digits '0' to '9' are guaranteed to be ordered and continuous, and the letters 'A' to 'Z' are ordered (but not necessarily contiguous).

The standard functions ORD and CHR converts from character to integer and back.

For example, in OXFORD Pascal:

$\text{ord}('A') = 65$
 $\text{chr}(36) = '\$'$

Also, $\text{succ}(x)$ gives the next character after x , and $\text{pred}(x)$ gives the character before x , for example:

$\text{succ}('3') = '4'$
 $\text{pred}('1') = '0'$

Note that in OXFORD Pascal $\text{succ}(\text{chr}(255))$ and $\text{pred}(\text{chr}(0))$ are undefined, and $\text{chr}(x)$ with x outside the range $0..255$ is not allowed.

3.2.4 User-defined (enumerated) types

These are usually defined by means of a type declaration (section 3.2) for example:

$\text{type day} = (\text{monday, tuesday, wednesday, thursday, friday, saturday, sunday});$

$\text{colour} = (\text{RED, GREEN, BLUE});$

The data type "day" then has seven ordered values represented by the identifiers MONDAY , TUESDAY , etc.

The type "colour" has three values. The function ORD , SUCC and PRED may be used on these types (see the previous section). For example.

$\text{succ}(\text{wednesday})$	$= \text{thursday}$
$\text{pred}(\text{green})$	$= \text{red}$
$\text{ord}(\text{monday})$	$= 0$
$\text{ord}(\text{green})$	$= 1$
$\text{ord}(\text{sunday})$	$= 6$

3.2.5 Subrange types

The user may define subranges over any scalar type except REAL.

Examples:

$\text{type year} = 1970..1990;$
 $\text{weekday} = \text{monday}..\text{friday};$

These types have the same properties as their parent types, but often occupy less storage space, and values are checked at runtime to see that they fall in the required range. They also act as a convenient means of documentation.

3.2.6 Boolean

Boolean values in Pascal are represented by the standard identifiers TRUE and FALSE. In fact the data type Boolean may be thought of as resulting from the declaration:

```
type boolean = (false, true)
```

so true > false. The Boolean operators defined in Pascal are:

and	—	logical "and" operation
or	—	logical "or" operation
not	—	(unary operator) logical negation.

The relational operators

	<	less than
	>	greater than
	=	equal to
<=		less than or equal to
>=		greater than or equal to
<>		not equal to

may be used with any scalar data type (integer, real, Boolean, char, user-defined), and give a Boolean result. They may also be used to compare strings (section 5.)

3.2.7 Operator precedence

The relational operators

< > <= >= = <> in (see section 5)

have lowest precedence, followed by

+ - or

then

* / div mod and

and finally the unary operator

not

Evaluation is otherwise left to right, and can be changed by using parentheses. Particular care should be taken with expressions like :

(x > 3) and (y = 2)

This would be illegal if the parentheses were omitted.

3.2.8 Summary of arithmetic and conversion functions

Function	Parameter	Result	Meaning
abs (x)	integer	integer	absolute value
abs (x)	real	real	absolute value
sqr (x)	integer	integer	square
sqr (x)	real	real	square
sqrt (x)	real or integer	real	square root (x >= 0)
ln (x)	real or integer	real	natural logarithm (x > 0)
exp (x)	real or integer	real	e raised to the xth power
sin (x)	real or integer	real	sine (x in radians)
cos (x)	real or integer	real	cosine (x in radians)
arctan (x)	real or integer	real	arctangent (0 to PI radians)
trunc (x)	real	integer	convert real to integer by truncation towards zero
round (x)	real	integer	converts real to integer by rounding
chr (x)	integer	char	convert ASCII value
odd (x)	integer	Boolean	TRUE if x is odd
ord (x)	scalar *	integer	position within a data type
pred (x)	scalar *	scalar	preceding value in a data type
succ (x)	scalar *	scalar	next value in a data type

(* can't be real)

The function "x to the power of y" may be calculated (for x > 0) using the expression :
exp (y * ln (x))

3.3. PASCAL DECLARATIONS AND STATEMENTS

3.3.1 Pascal Programs

A Pascal program takes the form:

```
program header
label declaration
constant declaration part
type declaration part
variable declaration part
function and procedure declarations
BEGIN
executable statements
END.
```

The declarations are all optional. Label declarations are discussed in III 3.12, functions and procedures in III 6.

The program header is optional in OXFORD Pascal. If it is included it consists of the keyword PROGRAM followed by a name (which can be any valid identifier) followed by a list of identifiers in brackets, for example:

```
program joe (input, output);
```

"Input" and "Output" are external files used by the program "joe". The header is terminated by a semicolon.

The final full stop after the program "end" is always required.

3.3.1.1 Constant declarations

These are used to assign values to identifiers which will not change throughout the program. They facilitate modifications to the program and provide a means of documentation.

The keyword "const" is followed by one or more declarations of the form

```
identifier = value;
```

"value" may be a signed or unsigned integer, real, a Boolean, character, string, a member of an enumerated type or a previously defined constant identifier.

Examples:

```
const      message      = 'hi there!';
           ch            = '$';
           PI            = 3.14;
           MINUSPI       = -PI
```

3.3.2 Type declarations

These are used to make an identifier synonymous with a given data type. The keyword "type" is followed by one or more declarations:

```
identifier = datatype;
```

Examples:

```
type      suit          = (SPADES, HEARTS, DIAMONDS, CLUBS);
           int           = integer;
           byte          = 0..255;
```

3.3.3 Variable declarations

In Pascal all variables must be declared explicitly. This is sometimes annoying but makes the programmer's intention clearer and helps the compiler to detect errors.

The word "var" is followed by one or more declarations:

```
identifier list: datatype;
```

Examples:

```
type day = (monday, tuesday, wednesday, thursday, friday);
var x,y:real;
i: integer;
switch : Boolean;
today, tomorrow, payday: day;
favouritecolour : (BLUE, RED, GREEN, PINK);
date : 1970..1990;
```

The variables denoted by these identifiers can then take any of the allowed values for the corresponding data type.

3.3.4 Executable statements

The executable part of a Pascal program enclosed by the keywords BEGIN and END, consists of zero or more sequentially executed statements separated by semicolons. Redundant semicolons are always accepted and generate no code. There is no need for any correspondence between the logical structure of statements and their physical layout. Well formatted programs with one statement per line are easier to read, however.

3.3.5 Assignment statements

The form of this statement is:

```
variable := expression
```

where the left and right hand sides must have compatible data types. This means that they must arise from the same type identifier, or be declared as variables in the same declaration. Exceptions are if the variable type is a subrange of the expression type, or they are sets with compatible base types, or if the left hand side is real and the right handside is integer.

The value of the variable is set to the value of the expression, and future references to the variable will yield this value.

Examples:

```
x := 3/sqrt(36)
y := x+4
y := y-2
x and y are
"real" variables.
x is set to 0.5
y is set to 4.5
y is set to 2.5
```

3.3.6 Compound statements

The construction:

```
BEGIN
Sequence of statements separated by semicolons
END
```

behaves as a single statement, which when executed causes the execution of all the enclosed statements in sequence.

3.3.7 "If" statements

The statement "IF Boolean expression THEN statement 1" causes statement 1 to be executed only if the expression is TRUE. Alternatively, "IF Boolean expression THEN statement 1 ELSE statement 2" causes statement 2 to be executed instead if the expression is FALSE.

IMPORTANT – no semicolon may be placed before the ELSE.

Statement 1 and statement 2 can be any Pascal statement, including another IF statement. For example: if x then if y then s1 else s2 – is taken to mean:

```
if x then
begin
if y then s1 else s2
end
```

3.3.8 "Repeat" statement

```
REPEAT
sequence of statements separated by semicolons
UNTIL Boolean expression
```

causes the sequence to be executed repeatedly (at least once) until the expression evaluates to TRUE when it is checked at the end of a loop.

3.3.9 "While" statement

WHILE Boolean expression DO statement 1

Statement 1 is repeated zero or more times until the expression turns out to be FALSE.

3.3.10 "For" statement

FOR variable := e1 TO e2 DO statement 1

The variable can be any scalar type except real. e1 and e2 are expressions of the same type as the variable. Statement 1 is executed exactly $\text{ord}(e2) - \text{ord}(e1) + 1$ times (zero times if $e2 < e1$). On successive loops the value of the variable is e1, succ(e1), succ(succ(e1)), ..., e2

An alternative form is:

FOR variable := e2 DOWNT0 e1 DO statement 1

Where statement 1 is executed with successively decreasing values of the variable.

Statement 1 should not try to alter the variable, as in:

```
for i := 1 to 10 do i := i + 1 (* WRONG *).
```

Structure members (section 5) can't be used as control variables in FOR loops.

Also, control variables must be local to the current block (section 6.4).

3.3.11 "Case" statement

```
CASE expression OF
constant list : statement;
constant list : statement;
:
:
constant list : statement;
END
```

A redundant semicolon may be included before the END, as shown.

Each constant list consists of one or more constants (which must be the same data type as the case expression), separated by commas. The case expression must be a scalar type (and can't be real). Each label in the case statement should be unique, and indicates that the statement it prefixes is the one to be executed if the case expression has that value. if no case labels match the expression value when the case statement is executed, a CASE ERROR occurs.

WARNING – Case statements with a wide spread of values should be avoided, for example:

```
Case n of
1: statement 1;
44255: statement 2
end
```

This will generate a large jump table in memory with null entries for all the intermediate values (2,3 etc.). Generally, case statements are an efficient way of choosing one of many similar statements to execute.

3.3.12 "Goto" statement

Pascal statements may be prefixed by a label thus:

```
label : statement
```

The label is an unsigned integer which should differ from all other labels in the first 8 digits in OXFORD Pascal (4 digits in standard Pascal). Control can then be transferred to this statement from another part of the program by means of the "goto" statements.

```
GOTO label
```

All labels must be declared before use (see below).

The effects of jumping into a structured statement (FOR, WHILE, REPEAT, IF, CASE, WITH) or into a function or procedure is undefined.

The use of GOTO's is not recommended if it can be avoided, since programs quickly become unreadable and error detection becomes very difficult.

Jumping to an undefined (as against undeclared) label is signalled as a runtime error in OXFORD Pascal.

GOTO's can be used to exit from nested functions and procedures.

Label declaration

This takes the form:

```
LABEL list of labels;
```

The labels are separated by commas.

3.4 INPUT AND OUTPUT OF TEXT

A file is a Pascal structured variable which (unlike an array) has no fixed size. Its elements are normally accessed sequentially and either reside on a disc or are associated with some physical I/O device such as the keyboard or display.

In this part we look mainly at textfiles, which are essentially files of characters (Pascal data type CHAR), but which give special treatment to the newline character. In particular the standard textfiles INPUT and OUTPUT, which are usually the keyboard and display, are discussed. Disc files are covered later in III.8 and III.9.

3.4.1 Outputting to textfiles

A textfile is a variable declared as type TEXT. Associated with any Pascal file *f* is a buffer variable *f↑* which is used in transferring data to and from the file. The standard procedure call:

```
write (f, ch)
```

is equivalent to:

```
f↑:= ch; put (f)
```

```
writeln (f)
```

sends a newline character (ASCII carriage return followed by a line feed) to the file *f*.

```
page (f)
```

sends a form-feed (or clears the screen in the case of the display).

3.4.2 Inputting from textfiles

```
get (f)
```

reads the next item from the file *f* into *f↑*.

```
read (f, ch)
```

for a character variable *ch* is equivalent to:

```
ch:= f↑; get (ch)
```

If the result of a GET is a newline character (a carriage return—linefeeds are ignored in textfiles), then *f↑* appears to contain a space and the standard function EOLN(*f*) is set to TRUE. Otherwise EOLN(*f*) is FALSE.

If the end of the file has been reached then get(*f*) will cause the standard function EOF(*f*) to become true, and *f↑* will be undefined. Doing a get(*f*) while EOF(*f*) is TRUE will cause an error.

```
readln(f)
```

skips to the start of the next line. It means:

```
begin while not eoln(f) do get (f); get(f) end
```


3.4.3 Reading other data types from textfiles

Syntax

read (f, variable list)

each variable in the list can be of type CHAR, INTEGER or REAL.

char : reads one character into the variable, as above.

integer : reads one character into the variable, as above.

integer : reads any valid (signed or unsigned) integer constant into the variable, skipping leading blanks and newlines.

real : reads any valid integer or real constant into the variable, skipping leading blanks and newlines.

f ↑ is set in each case to the next character after the data read.

3.4.4 Writing other data types to textfiles

Syntax:

write (f, expression list).

each expression can be of a type CHAR, REAL, BOOLEAN, INTEGER or a string, and may be qualified by a field width

expression : w

where w is a non-negative integer expression giving the total number of characters to write to the file.

Character or string:

Write sufficient spaces to give a total of w characters, then write the character or string. If w is too small then the string is truncated on the right. Default w = the size of the string.

Boolean:

As with string, but one of 'TRUE' or 'FALSE' is written. Default w=6.

Integer:

Write sufficient spaces first to give a total of w characters. Then write the number without leading zeros, preceded by a minus sign if it is negative. If w is too small print out the entire number with no spaces. Default w=7.

Real:

There are two formats:

(i) Floating point – write a sign character (space or '-') followed by a digit, followed by a decimal point, followed by enough digits to give a total of w characters, followed by a 4-character exponent. If w is too small, at least one digit is still printed after the decimal point. Default w=12.

(ii) Fixed point – the number of decimal places must be specified:

expression : w : d

Enough spaces are first printed to give a total of w characters, followed by a minus sign if negative, followed by a decimal point and d fractional digits, with roundings if necessary. If w is too small, no spaces are printed but the entire number is still output.

3.4.5 Abbreviations

write (f,...) is short for write (f,...);writeln (f)

read (f,...) is short for read (f,...);readln (f)

write (...) is short for write (output,...)

read (...) is short for read (input,...)

writeln is short for writeln (output)

readln is short for readln (input)

eoln is short for eoln (input)

eof is short for eof (input)

page is short for page (output)

3.4.6 Manipulating files

There is no problem in passing files as variable parameters. In OXFORD Pascal (but not in standard Pascal) assignment and passing as value parameters is also allowed. For example:

```
var sourcefile : text
:
:
begin
  sourcefile := input;
  :
  :
  read (sourcefile,x); (* reads from keyboard *)
```

3.5 STRUCTURED DATA TYPES

3.5.1 Arrays

The syntax of array types is:

ARRAY (indextype) OF element type.

Where "indextype" can be any scalar or subrange type except real. If indextype has values ranging from m to n, say, then this defines an array of $\text{ord}(n) - \text{ord}(m) + 1$ values of type "elementtype", which are referenced using the subscripts [m], [succ(m)], ..., [n].

Alternatively arrays can be accessed as a whole:

Examples:

```
var x,z : array [1..64] of integer;
```

```
y : array [0..3] of array [-4..21] of real;
```

```
begin
```

```
x [1] := 0;
```

```
x [5] := x [1] + 2;
```

```
y [3] [1] := 3.345;
```

```
z := x; (* Transfer whole array *)
```

"element type" may be any Pascal data type. N-dimensional arrays may be abbreviated as follows:

```
array [t1,t2,...,tn] of sometype
```

which is equivalent to:

```
array [t1] of array [t2] of ... array [tn] of sometype
```

example:

```
var x: array [1..7,4..9, boolean] of char;
```

reference to n-dimensional arrays may also be abbreviated,

```
x[i,7,false] := '$';
```

3.5.2 Sets

The syntax is:-

SET OF elementtype

where "elementtype" should be a scalar or subrange type, but not REAL. Sets are constructed from a collection of values in square brackets, for example:

```
var x : set of 0..127;
```

```
y : set of (RED, GREEN, BLUE);
```

```
begin
```

```
x := [1, sqrt(2), 6..74];
```

```
y := [BLUE, GREEN];
```

where 6..74 gives all the values between 6 and 74 inclusive. Set elements must have ordinal values between 0 and 127 inclusive. If their base types are compatible, then two sets are said to be compatible sets are:

*

+ and -

= <> <= >=

Intersection (highest precedence)

Union and difference

Equality, inequality and inclusion tests.

The IN operator tests membership of a set. The left hand side should be a scalar compatible with the sets base type. IN has the same precedence as the relational operators <>, = etc.

Examples:

Assuming

```
var x,y : set of (APPLES, PEARS, ORANGES, BANANAS, FIGS);
```

```
begin
```

```
x := [APPLES, PEARS, BANANAS];
```

```
y := [BANANAS, FIGS];
```

Then

```
x+y is [APPLES, PEARS, BANANAS, FIGS]
```

```
x-y is [APPLES, PEARS]
```

```
x*y is [BANANAS]
```

```
x=y, x<=y and x>=y are all false
```

```
x<>y is true
```

```
y<= [APPLES, FIGS, BANANAS] is true
```

```
y<=y is true
```

```
y>=y is true
```

```
y=[BANANAS, FIGS] is true
```

```
BANANAS IN y is true
```

```
ORANGES IN x+y is false
```

3.5.3 Records

The basic syntax is:

```
RECORD
identifier list : data type;
identifier list : data type;
:
identifier list : data type
END
```

An optional semicolon may be placed before the END. The fields may be accessed by the field name preceded by a dot, for example:

```
var x,y:record
    a,b:integer
    c:real
end;
begin
    x.b := -33;
    x.c := 9E-20;
    x.a := x.b+2;
```

Entire records may also be assigned:

```
x := y;
```

Several different record definitions may be combined using the following syntax:

```
RECORD
any field common to all variants
CASE identifier:datatype OF
    constant list : (field list);
    constant list : (field list);
:
:
    constant list : (field list)
END
```

Again there can be a redundant semicolon before the END. The variant "field lists" may themselves contain nested variants, for example:

```
type date = record
    year : integer;
    month : (JAN,FEB,MAR,APR,MAY,JUN,JLY,AUG,SEP,OCT,NOV,DEC);
    day : 1..31;
end;
```

```
person = record
    name: packed array [1..30] of char;
    birthday : date;
    case status: (EMPLOYED,UNEMPLOYED,RETIRED,STUDENT) of
        UNEMPLOYED: (registered : date);
        EMPLOYED : (case selfemployed : boolean of
            true : (numberofemployees:integer);
            false : (employer: packed array [1..30] of char;
                dateemployed : date))
    end;
```

```
var his:person;
begin
    his.name := 'Harry Johnson';
    his.birthday.year := 1938;
    his.birthday.month := DEC;
    his.birthday.day := 12;
    his.status := EMPLOYED;
    his.employer :=
        :
        etc.
    WITH statements have the effect of declaring the fields of a record as local
    variables for that statement. For example:

    with his birthday do
        begin
            month:=DEC;

            year := 1938;
            day := 12;
        end;
```

The record cannot however be referenced as a whole from inside the WITH statement.

WITH r1, r2, ..., m DO statement

is equivalent to:

WITH r1 DO WITH r2 DO ... WITH m DO statement

3.5.4 Packed Structures

Records, arrays, sets and files may be preceded by the word "packed". This is a command to the compiler to optimise storage space for that structure, possibly at the expense of speed in accessing individual components of the structure. In OXFORD Pascal, "packed" has little effect on speed, but may cut storage by half in arrays of enumerated values, characters and subranges (0..255 and less). The disadvantage is that packed array elements can't be used as VAR parameters to procedures or functions (but whole packed arrays can).

Packed arrays [1..n] of type CHAR are special in Pascal because they are considered to be string variables of length n.

Examples:

```
var x,y :packed array [1..4] of char;  
    z : packed array [1..10] of char;  
begin
```

```
    x := 'how';  
    y := 'when';  
    z := 'Hi there!';
```

y=x is false, y>x, y>=x and y<>x are true.
y>'what' is true, x<'why' is true.

But note:

x and z are incompatible (different lengths)
x and 'hello' are incompatible.
z and 'who' are incompatible.

3.5.5 Pack and Unpack (not available in resident mode)

Access to individual components of packed arrays may be costly, and the programmer is advised to pack or unpack a packed array in a single operation.

If U and P are array variables, for example:

```
type t=(some data type);  
var U : array [m..n] of t;  
    P:packed array [a..b] of t;
```

where (n-m) >= (b-a) then:

```
pack (U,i,P)
```

is equivalent to:

```
for j:=a to b do P [j] :=U [j-a+i]
```

and

```
unpack (P,U,i)
```

is equivalent to:

```
for j := a to b do U [j-a+i] := P[j]
```

3.6 FUNCTIONS AND PROCEDURES

3.6.1 Function and procedure definitions

The syntax for each definition is the same as the syntax for a program, except that a function or procedure header is used instead of a program header, and also a semicolon appears at the end instead of a full stop:

```
procedure or function header  
label declarations  
const definitions  
type definitions  
variable declarations  
procedure and function definitions  
BEGIN  
executable code for this procedure or function  
END;
```

Any number of procedures or functions may be defined in a program. The definitions should occur between the variable declarations and the main "BEGIN" of the program.

A procedure header has the form:

```
PROCEDURE procedurename;
```

or

```
PROCEDURE procedurename (formal parameter list);
```

A function header has the form:

```
FUNCTION functionname : datatype;
```

or

```
FUNCTION functionname (formal parameter list) : data type;
```

3.6.2 Procedure and function calls

Procedure calls are statements having the form:

```
procedurename
```

or

```
procedurename (parameters)
```

The effect is to execute any code between the BEGIN and the END of the procedure definition, and then return to continue the program normally, from the statement after the call.

Function calls are expressions which have the data type specified in the function header. To evaluate the function, any code between the BEGIN and the END of the definition is executed, and the value returned is the last value that was assigned to the function name. The value returned by a function must be a scalar or a pointer.

Examples:

```
procedure x;
begin
  writeln('xxxxx')
end;
begin x;
  writeln('yyyyy');
  x
end.
```

Is equivalent to

```
begin
  writeln('xxxxx');
  writeln('yyyyy');
  writeln('xxxxx')
end.
```

The following example will set i to the value of 4:

```
var i : integer;
function xyz : integer;
begin
  xyz := 2;
  xyz := 4
end;
begin
  i := xyz
end.
```

3.6.3 Parameters

The usefulness of procedure and function calls can be extended by passing parameters. If these are used they must correspond in number, position and type with the formal (dummy) parameters in the definition.

The formal parameter list contains one or more parts separated by semicolons. Each part has one of the forms:

identifier list : datatype
VAR identifier list : datatype
FUNCTION identifier list : datatype
PROCEDURE identifier list

These correspond to four different classes of parameters, identifiers, variables, FUNCTION and PROCEDURE parameters which are substituted with expression values, variables, function and procedure names respectively when the function or procedure is called.

Examples:

```
const SIZE = 20;
type vec = array [1..SIZE] of integer;
var v:vec ; i:integer;
function tan (x:real):real;
begin
  tan := sin (x)/cos(x)
end;
procedure zero (var a:vec);
begin
  for i:= 1 to SIZE do a[i] := 0
end;
function square (x:integer):integer;
begin
  square := sqr(x)
end;
function sigma (function f:integer; n,m :integer):integer;
var sum, i:integer;
begin
  sum:=0;
  for i:= n to m do sum:=sum+f(i);
  sigma:=sum;
end;
```

Given the above definitions

tan (0.5) would give the tangent of 0.5 radians
(sin (0.5)/cos(0.5))

zero (v) would set the array v to be all zeros.

Note that passing large arrays (and records) as VAR parameters is a good idea, because the computer does not then have to copy the array.

sigma (square, 1, 20)

evaluates 1+4+9+16+...+400.

OXFORD Pascal (and many other Pascal systems) will not let you pass standard function and procedure names as parameters, hence the need for the function "square".

WARNING – Functions and procedures passed as parameters can themselves only have value parameters, and these are not checked.

So:

```
procedure X(a:real);
begin
:
end;
procedure y(procedure b);
begin
    b(4)
end;
:
begin
y(x)
```

will lead to disaster because x expects a real and gets an integer parameter (4).

3.6.4 Local declarations

Any variables, constants, labels, types, procedures and functions declared within a procedure or function are local to that procedure or function and cannot be referred to from outside it.

"Global" identifiers defined outside a function or procedure may also be referenced inside it, unless they have been redefined by local definitions.

Examples:

```
program example;
var i:integer; (* may be referenced by main prog, P1 and P2 *)
    j:real; (* may be referenced by main prog and P3 *)
    k:boolean; (* may be referenced anywhere *)
procedure P1;
    var j:integer; (* may be referenced by P1 and P2 only *)
    procedure P2;
    var m:char; (* may be referenced by P2 only *)
    begin
    :
    end;
begin
:
end; (* of P1 *)
procedure P3;
const i:=49; (* may be referenced by P3 only *)
:
```

P1 and P3 may be called from anywhere.

P2 may be called from P1 or P2.

3.6.5 Recursion and forward references

Functions and procedures can call themselves recursively:

```
function factorial(x : integer):integer;
begin
    if x=0 then factorial := 1
    else factorial := factorial (x-1)*x
end;
```

factorial (4) gives $4*3*2*1 = 24$

Sometimes it is helpful for a procedure to be able to call another procedure before the procedure being called is defined. The undefined procedure must previously have been declared with name and parameter list, together with "forward" – see example. The parameter list is not required on subsequent declaration of the procedure.

Example:

```
procedure x(parameters for x); forward;
procedure y(parameters for y);
begin
    (* calls x *)
end;
procedure x;
begin
    (* calls y *)
end;
```

x and y call one another (they are "mutually recursive").

3.7 DYNAMIC STORAGE AND POINTERS

3.7.1 Pointers

Variables of a pointer type take as values the memory address of other variables. This can be used in Pascal to create variables as required while the program is running, since the compiler does not need to know the memory address in advance if it can be stores in a pointer. The syntax of a pointer type is:

type pointed to

where "type pointed to" is an identifier which is the name of some data type (which could be declared later, allowing recursive definitions such as linked lists and trees).

Examples:

```
type treepointer = tree;
  tree = record
    leftbranch, rightbranch : treepointer;
    data : sometype;
  end;
var oak : tree;
  p : integer;
```

The only way of giving a pointer a value in standard Pascal is to assign it the value "nil" (which is guaranteed to point to no variable) or to use the procedure "NEW".

In OXFORD Pascal, "nil" is the address 0000.

Pointers can, once assigned a value, be tested for quality (<> and =).

3.7.2 "New" and "dispose"

NEW allocates a new variable from the available storage (if any) and stores a pointer to it in the specified variable.

The variable created may then be referenced by the pointer variable followed by a.

DISPOSE destroys the variable pointed to by the specified pointer and makes the storage available for other use. Of course you must be sure that the variable being DISPOSED is never referenced again.

Examples:

```
var p: real;
begin
  new (p);
  p := 103.7;
  write (p * p * p);
  dispose (p)
end.
```

Would print the cube of 103.7 and then destroy the space used to store it. P means the variable whose address is in p.

3.8. DISK FILES

(sections 8.1 to 8.4 do not apply to resident mode)

3.8.1 Declarations

Disk files are declared as Pascal variables of type "file of X" where X is the base type of the file, and can be any structures or unstructured data type. For example:

```
type patient = record
  name : packed array [1..20] of char;
  wordnumber : integer;
end;
var f: file of integer;
  g,h: file of patient;
```

Every file f declared in Pascal has an associated buffer variable f↑ whose type is the base type of the file. Disk files can also be textfiles, for example:

```
var f1, f2 : text; (see section 4.1)
```

3.8.2 Sequential writing

Before they can be read or written, disk files must be opened using one of the standard procedures RESET and REWRITE. Up to 5 sequential disk files may be open at any time.

rewrite (f)

creates an empty file which is then open for sequential writing. The end-of-file eof(f) will return TRUE in this mode. The call put(f) writes the data in the file buffer (the variable f↑) to the file.

The sequence:

```
begin f↑ := expr; put (f) end
```

may be abbreviated to:

```
write (f,expr)
```

IMPORTANT NOTE – in OXFORD Pascal, assignments should not be made to the buffer variable f↑ before a reset(f) or rewrite (f) has been done.

3.8.3 Sequential reading

The procedure call:

```
reset (f)
```

opens the file f for sequential reading. f must previously have been written by a REWRITE command, otherwise the error message FILE DOES NOT EXIST will be printed. The first record in the file will be placed in the variable f↑. (Or if f is empty, f↑ will be undefined and eof(f) will be true).

Successive records can be read into the buffer variable `f↑` by the procedure call:

```
get(f)
read(f,x) is equivalent to x:=f↑;get(f)
```

The function `eof(f)` returns TRUE when there are no more records in the file. Attempt to read past an end-of-file will cause an error.

As an example the following program writes a file containing the numbers 1 to 10, and then reads them back displaying them on the 64 screen:

```
var i: integer;
    testfile: file of integer;
begin
  rewrite(testfile);
  for i:= 1 to 10 do write(testfile, i);
  reset(testfile);
  while not eof(testfile) do
    begin
      read(testfile, i);
      writeln(i)
    end
  end.
end.
```

3.8.4 External files

The files described above are "internal" files, in other words temporary files which are normally deleted when the program (or procedure or function) in which they are defined finishes. Permanent diskette files may be created and/or accessed by giving a filename parameter to RESET or REWRITE. (The parameter may be either a string constant or a string variable). This is an extension to standard Pascal allowing specification of filenames, which can be useful in interactive programs.

Note that the filename cannot contain any imbedded spaces. If the filename is a string variable, it should be terminated by at least one space.

Examples:

```
var fname: packed array [1..15] of char;
    f, g: file of sometype;
begin
  reset(f,'datafile')
  fname := 'O:TEMP,HEX  '; (* drive must be specified *)
  rewrite(g,fname);
```

3.8.5 Reading and writing from other devices

Any device on the IEEE bus may be accessed by using RESET or REWRITE with the syntax:

```
reset(f, devicenumber,secondaryaddress)
or reset(f, devicenumber,secondaryaddress,filename)
```

where device number and secondary address are integer expressions. This syntax can also be used in resident mode.

```
var printer:text;
begin
  rewrite(printer,4,0);
  writeln(printer,'Message with UPPER case!');
```

The rewrite command may be used to send commands to the floppy disk unit, for example:

```
const DISK = 8; (* disk unit physical device # *)
      CC = 15; (* command channel secondary address *)
var f: text;
begin
  rewrite(f, disk, cc, '11'); (* Initialize drive 0 *)
  rewrite(f, disk, cc, 'RO:NEWNAME=OLDNAME')
    (* Rename a file *)
  rewrite(f, disk, cc, 'C1!COPY=0;FILE1,0:FILE2');
    (* Copy disk files *)
```

3.8.6 CLOSE command

This command is an extension to standard Pascal. It may be used to explicitly close a file (without resetting or rewriting) if required. The syntax is:

```
close(f)
```

Disk textfile example

The following example program prompts the user for a idsk file name, and then outputs an upper-and-lower-case textfile to the 64 printer.

```
program printfile;
var fname: packed array [1..80] of char;
    ch: char;
    f, printer: text;
begin
  writeln;
  writeln('Filename ? ');
  read(f, fname);
  reset(f, fname);
  rewrite(printer, 4,0);
  while not eof(f) do
    begin
      while not eoln(f) do
        begin
          read(f, ch);
          write(printer, ch);
        end;
      readln(f);
      writeln(printer);
    end
  end.
end.
```

3.9 EXTENSIONS TO STANDARD PASCAL

The features described in this section are specific to OXFORD Pascal and might not be implemented on other systems.

3.9.1 Hexadecimal constants

These are introduced by the symbol \$ (for integer constants) or a backslash (for character constants).

Their main application is probably in machine language and I/O interfacing

Examples:

```
const portA=$e84f;
      linefeed= a;
var  chardata:char;
begin
.
.
.
      chardata:=linefeed; (* linefeed is a constant of type CHAR *)
      poke (portA, $3f);
```

(writes the data 3f hex to an imaginary VIA port "A" mapped at hex memory address E84F)

3.9.2 Memory VDU and port access

The standard functions/procedures PEEK, POKE, ORIGIN, GETKEY and VDU are provided for this purpose.

peek (x:integer):0..255

is a function which gives the contents of the physical memory location x, while the procedure:

poke (x:integer; y:0..255)

is used to change the contents of location x to the byte y. Poke should, of course, be used with great care to avoid corrupting your program.

origin(x : sometype;y :integer)

sets the pointer x to point at the physical memory location y. x can be any pointer type. This should be used with care (see section 10).

The procedure VDU (x,y :integer; c :char) stores the character c in the VDU memory row x, column y.

Finally, the function

getkey:char

returns a character read directly from the keyboard port. Chr(0) is returned if no character is ready.

Examples:

```
var x:0..255;
begin poke($014c, $33);    stores the byte 33 (hex) at address 14c (hex)
x:= peek(47);              sets x to the contents of decimal memory address 47
page;                     clears the VDU screen
VDU(0,3,'?');              writes a question mark to the VDU row 0, column 3
while getkey=chr(0)do;     waits for someone to press a key
```

3.9.3 Added Commands for Oxford Pascal V1.0

SOUND COMMANDS.

There are three sound commands available as pre-declared procedures in Oxford Pascal. These allow use of all three voices and envelope generators.

PROCEDURE NAME	ENVEL (V,A,D,S,R);		
PARAMETER		TYPE	RANGE
V is the voice number.		Integer	1 - 3
A is the attack rate.		Integer	0 - 15
D is the decay rate.		Integer	0 - 15
S is the sustain rate.		Integer	0 - 15
R is the release rate.		Integer	0 - 15

PROCEDURE NAME	VOICE (V, F, W, D);		
PARAMETER		TYPE	RANGE
V is the voice number.		Integer	1 - 3
F is the frequency.		Integer	0 - 65535
W is the wave type.		Integer	0 - 3
D is the duration.		Integer	0 - 65535

N.B.

The wave types are triangle, sawtooth, pulse, noise. The pulse length of the pulse signal has been preset to an equal mark space ratio.

The frequency is as defined in the COMMODORE 64 Handbook and is determined by the formula $F_{out} = (F * 0.059604645) \text{ Hz}$

The duration is a constant used in an internal delay loop and denotes the delay from the signal reaching its sustain level till the start of the release cycle. Realistic effects can be obtained by making the duration a factor of the sustain level multiplied by some constant.

The ENVEL command must precede the VOICE command since the latter acts as the trigger for the voice.

PROCEDURE NAME	VOLUME (L);		
PARAMETER		TYPE	RANGE
L is the volume level.		Integer	0 - 15

The volume command controls the master volume of the SID chip in the 64. It can be set to any level at any time.

GRAPHICS COMMANDS

Oxford Pascal provides various pre-declared procedures and functions for use of the graphics on the 64. These can be split into commands for normal screen use, and hi-resolution commands.

Normal screen commands.

Set the border colour.

BORDER(C);

Set the screen colour.

SCREEN(C);

Set the colour for the text to print in.

PEN(C);

C is an integer in the range 0 – 15 and determines the colour as defined in the COMMODORE 64 HANDBOOK.

Hi-Resolution commands.

Set the background colour to plot on.

PAPER(C);

Set the colour to plot in.

INK(C);

Hi-Resolution screen switch. (0=OFF / 1=ON).

HIRES(C);

Examine a point. (0 = OFF / 1 = ON)

P:=EXAMINE(X,Y);

Multi-purpose hi-resolution command.

PLOT(F,X,Y,X1,Y1);

All parameters are integers.

F = 0 clear background to paper colour.

F = 1 clear all points on hi-resolution screen.

F = 2 plot a line from X,Y to X1,Y1

F = 3 clear a line from X,Y to X1,Y1

F = 4 fill an area around point X,Y till nearest boundaries

F = 5 clear an area around point X,Y till nearest boundaries

To plot or clear a point make X1,Y1 = X,Y

Create a Text window on Hi-Resolution screen starting at top of screen and terminating at line U. WINDOW(U);

3.9.4 Hexadecimal input and output

The procedure WRHEX and WRHEX2, and the function RDHEX are provided.

wrhex (f:text; x:integer)

writes x as four hex. digits on the textfile f.

wrhex2 (f:text; x:0..255)

writes the byte x as two hex. digits.

Examples:

rewrite (printer,4,0);

wrhex (printer, -1); wrhex2 (output, 3)

prints FFFF on the printer and 03 on the 64 screen.

The function,

rdhex (f:text):integer

reads a 16 bit value from the file f, skipping any leading blanks and discarding all but the last four digits read.

3.9.5 Bit manipulation

ANDB, ORB, XORB, NOTB, SHL, and SHR are functions operating on integers but treated as 16 bit logical data. The first four do bitwise AND, inclusive OR, exclusive OR and 1's complement.

SHL(x,y) shifts x left by y bits(zeros are shifted in)

SHR(x,y) shifts x rights by y bits

SHL(x,-y) is equivalent to SHR(X,y)

examples:

andb(\$fff0,\$00ff)=\$00f0

orb(\$ff00,\$000f)=\$ff0f

xorb(\$ff00,\$0ff0)=\$f0f0

notb(\$f0f0)=\$0f0f

shl(4,4)=\$40

shl(4,-1)=\$4000

shr(4,0)=shr(4,0)=4

shr(\$4444,4)=\$444

3.9.6 Catching I/O errors

Occasionally it is necessary for a program to protect itself against unexpected termination due to invalid input.

The procedure call:

iotrap(false)

turns off PASCAL error messages for real and integer read operations and disk I/O:

iotrap(true)

turns checking back on again. After each integer or floating point or hex read operation the function IOERROR may be used giving an integer error number.

ioerror= 0-No error
2-Integer read error
10-Floating point read error
etc. see (section II.8 for a complete list of I/O runtime errors).

3.9.7. Keyboard interrupts

The calls:

restore(true)

restore(false)

enable and disable the restore key respectively.

The default is restore (true).

3.9.8. Random Number Generator

The function random :0..255 gives a random no. between 0 and 255. A pseudo-random generating sequence is used but this is initialised by timing all keyboard inputs and is also "kicked" frequently by the PASCAL interpreter.

The construction

random+(random mod 128)*256

generates a random no. between 0 and MAXINT, while

random mod + 1

generates an (almost) random no. in the range 1..n if n is not too large.

3.9.9. Underscore

The character '_' (CBM Key and @ Key) is allowed as a letter in identifiers giving improved readability.

3.9.10 The 64 internal clock

The clock may be examined by using the three functions :

hours : integer

minutes : integer

seconds : integer

and may be set using the procedure settime (h,m,s : integer).

Example:

settime (12,47,00);

Sets the clock to 47 minutes past midday and

writeln (hours, ':', minutes, ':', seconds);

would print:

12: 47: 0

3.9.11 Input of String Variables

String variables (ie packed arrays [1..n] of char) may be read from textfiles in a similar manner to characters, integers and reals. Any leading spaces or newlines are first skipped, then an entire line of characters is read from the file into the string variable. If the string is too long, it is truncated on the right, if it is too short it is padded out with spaces.

A major application is for inputting file names from the console.

3.9.12 Program chaining (disk mode only)

The OXFORD Pascal command :

chain (filename)

stops execution of the current program and invokes the program named. The value of GLOBAL variables will be preserved only if declarations are identical in the old and new programs. All files are closed.

The filename can be either a string or a string variable. (If a string variable, at least one space must be used as terminator).

When used under the EX command, a ".obj" extension is implied.

Example:

file "prog1" (object code in "prog1.obj"):

```
begin
  writeln('First program');
  chain('Prog2')
end.
```

file "prog2" (object code in "prog2.obj"):

```
begin
  writeln('Second program');
  chain('Prog1')
end.
```

The command

ex prog1

would cause the following to be printed:

```
First program
Second program
First program
Second program
:
```

until the stop key is pressed.

Program chaining is a useful technique for splitting up large programs, or for menu-driven applications.

3.10 OXFORD PASCAL INTERFACE GUIDE

The purpose of this section is to provide all the necessary information to write 6502 machine language subroutines for OXFORD Pascal programs.

3.10.1 Assembly language format

Assembly language routines are declared as Pascal functions or procedures but the body is replaced by the word "extern" followed by an integer constant (the routine address). Any parameters are passed on the stack and should be removed by the assembly language routine. The routine should also push a return value on the stack if it is declared as a function. The best way to describe this is by example, so here is a simple function to add two integers:

```
program test;
function addxy (x,y:integer):integer;
  extern $C000;
begin
  write (addxy(3,4))
end.
```

This should result in the output:

7

Provided that the assembly language routine is correctly located at memory address C000h:

```
sptr      = $2A ; Pascal stack ptr
*         = $C000
addxy      clc
           ldy #0
           lda (sptr),y      ; low byte y
           ldy #2            ; low byte x
           adc (sptr),y
           sta (sptr),y      ; low byte result
           dey
           lda (sptr),y      ; hi byte y
           ldy #3
           adc (sptr),y      ; hi byte x
           sta (sptr),y      ; hi byte result
           clc
           lda sptr          ; pop Y, leave result
           adc #2
           sta sptr
           bcc addrts
           inc sptr+1
addrts     rts
```

Note: the top-of-memory pointer at locations \$37-38 should first be set to \$C000 or below to prevent Pascal from overwriting these locations.

3.10.2 Storage formats

All scalar and subrange types (except REAL), and pointers are passed as 16-bit words in the usual low-high format.

Reals are passed as 6 bytes; in 64 BASIC format.

```
loc n+5: unused
loc n+4: LS mantissa
loc n+3: .
loc n+2: .
loc n+1: MS mantissa
loc n   exponent
```

Arrays are stored row-by-row (the opposite to FORTRAN), the lowest element has the lowest address.

Arrays are byte-packed if their elements are scalars in the range 0..255 (eg. char), and "packed" was specified. In this case the size is always rounded up to an even number of bytes.

Records are stored with their fields in reverse order (first declared has highest address). Sets are passed as a 128-bit map, a "one" indicates membership. Odd and even bytes are reversed:

```
loc n+15:
loc n+14
:
loc n+1
loc n

bit 15 .... bit 8
bit 7 ... bit 0
```

```
bit 127 ... bit 120
bit 119 ... bit 112
```

IMPORTANT – pointers always point to the location above the highest byte used by the actual data. This also applies to VAR parameters, which are passed as addresses.

Example:

```
const VDUSIZE = 1000; (* 25 rows of 40 chars *)
type screen = packed array [1..VDUSIZE] of char;
var vduptr : screen;
begin
    origin (vduptr, $0400 + VDUSIZE)
    :
    :
```

This declares an array based on the 64 vdu address 0400h. vduptr[1] is the first vdu location.

ADDENDA

Directories

A disk directory may be obtained from the editor using the BASIC command:

```
LOAD "$0",8 for drive 0
LOAD "$1",8 for drive 1
```

Warning – this will overwrite any program text in memory.

LOCATE command

Be careful not to give the BASIC runnable file the same name as your source file (or any other Pascal file on the same disk) as the Commodore will not overwrite a file with one of a different type.

Compiling Errors

Error reporting in resident mode now produces the appropriate error message and not an error number. Be prepared to look several lines back in your program for an error as it may take this long to confirm an error. Also one error may produce a series of other errors which will disappear when the first error is corrected.

High Resolution Graphics – extra notes

1. Screen boundaries; The x, y coordinates range from 0-255 and 0-200 respectively. The origin 0, 0 is at the bottom left hand corner of the screen area. Any point plotted out of range of these bounds will not be performed.
2. HIRES (1) defaults at power up to a complete screen equivalent to the command WINDOW (25).
3. It is advised that a HIRES (0) statement is used before terminating a program using hi-res graphics, unless it is desired to leave a window showing a particular display. If a program exits or is stopped without this being done the command KILL can be typed from the editor (blind, if necessary). KILL disables the hi-res screen and removes the graphics wedge from the interrupt routine.
4. To enable a clean display in hi-res mode both functions 0 and 1 of the plot command should be used.
5. Oxford Pascal uses standard bitmap mode for high resolution graphics and therefore, whilst having access to 16 different colours, any character location on the screen can only have one foreground (ink) and one background (paper) colour. A character location consists of an 8*8 array of points. The effects of this are noticeable when two lines of different colours are made to cross when part of the line being crossed around the intersection changes to the colour of the line crossing it.

Demonstration graphics program

(* this should produce a circle filled with pretty patterns*)

```
var z,x,y,l: integer;
    a,b,c,r,p: real;
begin
  WINDOW(14);
  R:=40;
  Z:=1;
  C:=1;
  PAPER(10);
  INK(3);
  PLOT(0,0,0,0,0);
  PLOT(1,0,0,0,0);
  HIRES(1);
  for l:=0 to 359 do
  begin
    p:=c*l;
    a:=r*COS(p);
    b:=r*SIN(p);
    x:=ROUND(a);
    y:=ROUND(b);
    INK(Z);
    z:=z+1;
    PLOT(2,y+150,x+150,x+150,y+150);
  end;
end.
```

Demonstration sound program

(* this should make some interesting modulated sounds*)

```
var t: integer;
begin
  ENVEL(1,0,0,13,15);
  for t:=15 down to 0 do
  begin
    VOICE(1,1000*t,1,7000);
    VOICE(1,1000*t,0,7000);
    VOLUME(t);
  end;
end.
```

Useful Memory Locations

HEX	DEC	USE
C000	49152	Printer device number in editor mode – default 4
C001	49153	Printer type – default 1 0 = Pet or 64 type 1 = Ascii
C002	49154	Line feed flag – default auto line feed 0 = auto line feed disabled 1 = auto line feed



Limbic Systems U.K. Ltd.
Hensington Road, Woodstock, Oxford OX7 1JR, England
Telephone (0993) 812700

Limbic Systems Inc.
560 San Antonio Road, Suite 202, Palo Alto, Ca. 94306 U.S.A.
Telephone (415) 424-0168